
文件中的习惯用法 

 

此手冊中出現的习惯用法。 

 

习惯用法          說          明 

 

[expression list]  雙引號內的資料長度最大 2*1024 bytes; 

 

<ESC> 跳脫符號 (ASCII 27)，為一立即指令，不管印表機是否為

待機或忙碌中，均立即回覆印表機目前將態。 

 

~      (ASCII 126)，當印表機於待機狀態時可回傳印表機狀態。 

 

Space , Tab           當(ASCII 32)、(ASCII 9)字元出現在指令列時將被忽略。  

(目前 F/W 遇 Tab 會當機) 

“          (ASCII 34)，資料字串的開始與結尾判別符號。 

 

CR,LF (ASCII 13)、(ASCII 10)或(ASCII 10)可做為指令列的分段符號 

(目前 ASCII 10 即 Hex0A 無法單獨取代 0D0A 於指令末端使用, 將

請 RD 修改) 

 
 

NULL (ASCII 0)於一般資料陳述中並不支援，僅供給二維條

碼使用。 

 

\nnn (???) (ASCII 92)、nnn 為一數字，可用 000 至 255 的十

進位方式來表示字元。 

 

FF , EOF (???) (ASCII 12)、(ASCII 26) 僅可被使用於指令列的結

尾。 

 
 

注意事項：200 DPI: 1 mm = 8 dots  細明體 The的粗斜體字型表示注意事項。 

Set up and System Commands 
 SIZE 
 

功能：定義標籤紙的寬度及高度 

 

語法：(1) 英制系統 (英吋) 

          SIZE m,n 

      (2) 公制系統 (公釐) 

          SIZE m mm, n mm 
 

參   數      說          明 

m    定義標籤紙的寬度 (英吋或公釐) 

n    定義標籤紙的寬度 (英吋或公釐) 

 

注意事項:  

200 DPI: 1 mm = 8 dots 
300 DPI: 1mm = 12 dots 

在公制系統下, 在參數 m 與 “mm” 之間必須要有空格分隔 

 

最大寬度 102mm 104mm 106mm 108mm 

TTP-2410M  X   

TTP-248M  X   

TTP-246M    x 

TTP-246G    x 

TTP-344M   x  

TTP-245    x 

TDP-245    x 

TTP-245G    x 

TDP-245G    x 

TTP-343   x  

 
 

範例： 

(1) 英制系統 (英吋) 

SIZE 3.5, 3.00 

(2) 公制系統 (公釐) 

SIZE 100 mm, 100 mm 
 

 
 

其他參考項目： 

GAP, BLINE 

 GAP (Add GAP 0,0 to define the continuous label) 
 

功能：定義兩張標籤之間的間隙距離 

 

語法：(1) 英制系統 (英吋) 

          GAP m,n 

      (2) 公制系統 (公釐) 

          GAP m mm, n mm 
 

參   數      說          明 

m    定義標籤間隙高度 (英吋或公釐) 

                 0  m  1 (inch), 0  m  25.4 (mm) 

n    定義標籤間隙高度的補償值 (英吋或公釐) 

  n  label length (inch or mm) 

0,0              表示連續紙模式 

 

注意事項: 

在公制系統下, 在參數 m 與 “mm” 之間必須要有空格分隔 

另當標籤紙由 “黑標紙張” 換成 “間隙標籤” 時, 請先下達 “GAP” 的設 

定指令 

如：在 DOS 模式下 

    C:\>COPY CON LPT1 <Enter> 
        GAP 2 mm,0<Enter> 
        <Ctrl>+<Z><Enter> 

 

範例： 

一般標籤紙 

(1) 英制系統 (inch) 

GAP 0.12,0 

(2) 公制系統 (mm) 

GAP 3 mm,0 

(3) 連續紙模式 

GAP 0,0 
 

 
 
 
 

特殊標籤紙 

(1) 英制系統 (inch) 

GAP 0.30, 0.10 

(2) 公制系統 (mm) 

GAP 7.62 mm, 2.54 mm 
 

 
 
 

其他參考項目： 

SIZE, BLINE 

 BLINE 
 

功能：設定黑標高度及使用者定義標籤印完後標籤額外送出的長度 

 

語法：(1) 英制系統 (英吋) 

            BLINE m,n 

        (2) 公制系統 (公釐) 

             BLINE m mm, n mm 
 

 參   數        說          明 

 m     黑標高度 

      0.1  m  1 (inch), 2.54  m  25.4 (mm) 

 n     額外送出紙張長度 distance between label leading edge and 

the  

                     beginning of black line. 0  n  label length 

0,0                 連續紙 

 

注意事項： 

在公制系統下, 在參數 m 與 “mm” 之間必須要有空格分隔。另當標籤紙由“間隙標籤” 

換成 “黑標紙張” 時, 請先下達 “BLINE” 的設定指令 

如：在 DOS 模式下 

    C:\>COPY CON LPT1 <Enter> 
        BLINE 2 mm,0<Enter> 
        <Ctrl>+<Z><Enter> 

範例： 

(1) 英制系統 (英吋) 

BLINE 0.20,0.50 
 

(2) 公制系統 (公釐) 

BLINE 5.08 mm,12.7 mm 
 

 
其他參考項目： 

SIZE, GAP 

 OFFSET 
 

功能：定義標籤於列印完後額外推出的長度，特別是在使用自動剝紙器或裁刀功能時，可用來

調校標籤停止的位置，在列印下一張時印表機會將原先多推出或少推出的部份以回拉方

式補償回來 

    

語法：(1) 英制系統 (英吋) 

     OFFSET m 

      (2) 公制系統 (公釐) 

             OFFSET m mm 
 

參數           說          明 

m     偏移距離 (英吋或公釐) 

      -1 m 1(inch) 
       

警告：不適當的偏移值會導致出現 “卡紙” 的現象 

 

範例： 

(1) 英制系統 (英吋) 

OFFSET 0.5 
 

(2) 公制系統 (公釐) 

OFFSET 12.7 mm 
 
 

 
其他參考項目： 

SIZE, GAP, SET PEEL, SET CUTTER 

 SPEED  
 

功能：設定印表機的列印速度 

 

語法：SPEED n 

 

參     數   說          明 

n     每秒的列印速度，以英吋計算 

 

N 1 1.5 2 3 4 5 6 8 10 12 

TTP-2410M     x  x x x  

TTP-248M     x x x x   

TTP-246M   x x x x x    

TTP-246G   x x x x x    

TTP-344M   x x x      

TTP-245   x x x x     

TDP-245   x x x x     

TTP-245G   x x x x     

TDP-245G   x x x x     

TTP-343   x x       

 

 範例： 

SPEED 10 
 

其他參考項目：： 

DENSITY 

 DENSITY 
 

功能：設定印表機的列印濃度 

 

語法：DENSITY n 

 

參     數   說          明 

n     0~15 

0 表示最淡的濃度 is  

      15 表示最深的濃度 

 

    範例： 

DENSITY 7 
 

其他參考項目：： 

SPEED 

 DIRECTION and Mirror Image 
 

功能：設定列印方向及鏡射功能，此設定將被記錄於 EEPROM 

    

語法：DIRECTION n[,m] 

 

參     數   說          明 

n     0 或 1 請參考下方圖例 

m                   0: 正常列印 1: 以鏡射方式列印 

 
 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
  

範例： 

DIRECTION 0[,0] 
 

其他參考項目：： 

REFERENCE 

 REFERENCE 
 

功能：定義標籤紙上的相對於原點的參考點座標，請參考下方圖例 

 

 
 

 語法： 

REFERENCE x, y 
 

參       數   說          明 

x     水平座標，單位 “dot”  

y     垂直座標，單位 “dot” 

 

注意事項：200 DPI: 1 mm = 8 dots 

   300 DPI: 1 mm = 12 dots 
which is shown in dot. 

 範例： 

REFERENCE 10,10 
 

其他參考項目：： 

DIRECTION 

 COUNTRY 
 

功能：當搭配 KP-200 (選配) 鍵盤時，此項設定可應因不同的國家提供該國文字於 KP-200 的 

LCD 螢幕上 

 

語法：COUNTRY n 

 

參     數   說          明 

n     001: USA 
      002: Canadian-French 
      003: Spanish (Latin America) 
      031: Dutch 
      032: Belgian 
      033: French (France) 
      034: Spanish (Spain) 
      036: Hungarian 
      038: Yugoslavian 
      039: Italian 

      041: Switzerland 
      042: Slovak 
      044: United Kingdom 
      045: Danish 
      046: Swedish 
      047: Norwegian 
      048: Polish 
      049: Germany 
      055: Brazil 
      061: English (International) 
      351: Portuguese 
      358: Finnish 
 

 範例： 

COUNTRY 001 
 

其他參考項目：： 

CODEPAGE, ~!I 

 CODEPAGE 
 

功能：改變內建文字的字元表，以符合各國家的使用 

 

語法：CODEPAGE n 

 

參     數     說          明 

n               字元表可被區分為 7-bit 字元表及 8-bit 字元表 

7-bit code page name 
USA: USA 
BRI: British 
GER: Germany 
FRE: French 
DAN: Danish 
ITA: Italian 
SPA: Spanish 
SWE: Swedish 
SWI: Swiss 
 
8-bit code page number 
437: United States 
850: Multilingual 
852: Slavic 
860: Portuguese 
863: Canadian/French 
865: Nordic 
 

注意事項：7-bit 或 8-bit 的字元表使用取決於串列埠傳輸的資料長度. 

 

 範例： 

CODEPAGE 437 
 

其他參考項目：： 

COUNTRY, SET COM1, ~!I 

 CLS 
 

功能：清除 Image Buffer  

    

語法：CLS 

 

注意事項：此項指令必須置於 SIZE 指令之後 

 

範例： 

CLS 
 

其他參考項目：： 

SIZE, GAP, BLINE 

 FEED 
 

功能：將標籤紙向前推送指定的長度 

 

語法：FEED n 

 

參     數   說          明 

n     單位：dot 

      1  n  9999 
 

 範例： 

FEED 40 
 

注意事項：200 DPI: 1 mm = 8 dots 

     300 DPI: 1 mm = 12 dots 
 

其他參考項目：： 

BACKFEED, SIZE, GAP, BLINE, HOME, FORMFEED 

 BACKFEED 
 

功能：將標籤紙向後回拉指定的長度 

    

語法：BACKFEED n 

 

參     數   說          明 

n     單位：dot 

      1  n  9999 
 

警告：不當的回拉距離會造成 “卡紙” 或 “碳帶皺折” 的現象 

注意事項：200 DPI: 1 mm = 8 dots 

    300 DPI: 1 mm = 12 dots 
 

 範例： 

BACKFEED 40 
 

其他參考項目：： 

FEED, SIZE, GAP, BLINE, HOME, FORMFEED 

 FORMFEED 
 

功能：將標籤向前推送至下一張標籤的起始位置 

 

語法：FORMFEED 

 

範例： 

SIZE 4,2.5 
GAP 0 mm,0 
SPEED 4 
DENSITY 7 
DIRECTION 0 
OFFSET 0.00 
REFERENCE 0,0 
SET PEEL OFF 
SET CUTTER OFF 
SET COUNTER @0 +1 
@0="000001" 
FORMFEED 
CLS 
BOX 1,1,360,65,12 
TEXT 25,25,"3",0,1,1,"FORMFEED COMMAND TEST" 
TEXT 25,80,"3",0,1,1,@0 
PRINT 3,1 
 

其他參考項目：： 

FEED, SIZE, GAP, BLINE, HOME, BACKFEED 

 HOME  
 

功能：在使用含有間隙或黑標的標籤紙時，若不能確定第一張標籤紙是否在正確列印位置時，

此指令可將標籤紙向前推送至下一張標籤紙的起點開始列印。 

 

語法：HOME 

 

範例： 

SIZE 4,2.5 
GAP 2 mm,0 
SPEED 4 
DENSITY 7 
DIRECTION 0 
OFFSET 0.00 
REFERENCE 0,0 
SET PEEL OFF 
SET CUTTER OFF 
SET COUNTER @0 +1 
@0="000001" 
HOME 
CLS 
BOX 1,1,360,65,12 
TEXT 25,25,"3",0,1,1,"HOME COMMAND TEST" 
TEXT 25,80,"3",0,1,1,@0 
PRINT 3,1 

 
其他參考項目： 

FEED, SIZE, GAP, BLINE, FORMFEED 

 PRINT 
 

功能：將存於 Image Buffer 的標籤印出 

    

語法：PRINT m [,n] 

 

參     數   說          明 

m     The quantity of 列印張數 sets 

      1  m  999999999 

若 m=-1, 重印最後一張標籤. 

 

n 每張標籤需重覆列印的張數 

      1  n  999999999 
  

 範例： 

SIZE 60 mm, 20 mm 
SET COUNTER @1 1 
@1=”0001” 
CLS 
TEXT 10,10,”3”,0,1,1,@1 
PRINT 3,2 
PRINT –1,2 

 
 
 
 
 
 

        
其他參考項目： 

SET COUNTER, INPUT, DOWNLOAD 

 SOUND 
 

功能：控制印表機蜂鳴器發聲的頻率，共有 10 階，並可由 “interval” 參數設定音與音的間隔

時間 

    

語法：SOUND level,interval 

 

參     數   說         明 

level     音階：0~9 

interval    間隔時間: 1~4095 

 
範例： 

SOUND 5,200 
SOUND 3,200 
SOUND 3,200 
SOUND 4,200 
SOUND 2,200 
SOUND 2,200 
SOUND 1,200 
SOUND 2,200 
SOUND 3,200 
SOUND 4,200 
SOUND 5,200 

 CUT 
 

功能：不做回拉動作，直接執行裁切動作 

 

語法：CUT 

 

範例： 

SIZE 3,3 
GAP 0 mm,0 
CLS 
DIRECTION 0 
REFERENCE 0,0 
OFFSET 0.00 mm 
SET CUTTER OFF 
SET TEAR OFF 
BOX 0,0,866,866,5 
TEXT 100,100,"5",0,1,1,"FEED & CUT" 
TEXT 100,200,"5",0,1,1,"300 DPI" 
PRINT 1,1 
FEED 260 
CUT 

 
其他參考項目： 

SET CUTTER, SET BACK, SET PARTITAL_CUTTER 

 LIMITFEED 
 

功能：限定定間隙校正執行的最大長度，若在此長度範圍內無法測得間隙存在，則將感應器模

式定在連續紙模式下 

 

   語法：(1) 英制系統 (英吋) 

LIMITFEED n  

 (2) 公制系統 (公釐) 

LIMITFEED n mm 
 

參     數   說          明 

n        間隙校正的最大長度 

 

範例： 

     LIMITFEED 10 
 
註： 

1. 此項設定會被記錄於 EEPROM 中 

2. 初始化的預設最大長度為 10英吋 

3. 在公制系統中，參數 m 與單位 “mm” 之間必須有空格分隔 

 SELFTEST 
 

功能：不經自測動作，直接將印表機訊息印在標籤紙上。 

    

語法： 

SELFTEST 
 

   範例： 

SELFTEST 

 CONFIG 
 

功能：經 RS-232 回傳有關印表機設定訊息 

    

語法： 

CONFIG 
 

   範例： 

CONFIG 

Label Formatting Commands 
 BAR 
 

功能：畫線或長條物 

    

語法：BAR x, y, width, height 

 

參     數   說     明 

x     線條左上角 X 座標，單位 dot 

y     線條左上角 Y 座標，單位 dot 

width    線條寬度，單位 dot 

height    線條高度，單位 dot 

 

注意事項：200 DPI: 1 mm = 8 dots 

      300 DPI: 1 mm = 12 dots 

        為保護電源供應器的正常運作，建議於列印 4 吋寬的線條時，最大高度不超過 12 

dots，否則將有可能損壞電源供應器及影響列印品質 

        另為保護印字頭的正常使用，對於桌上型機器與工業用機器單一點行最大的列印比率
分別為 20%及 30%。以 200dpi 機器為例，其單一點行為 864 dots，則桌上型機種
有效的列印範圍為 172 dots，工業型則為 259 dots。 

 

 範例： 

SIZE 4,2.5 
GAP 0,0 
SPEED 6 
DENSITY 8 
DIRECTION 0 
CLS 
BAR 100, 100, 300, 200 
PRINT 1,1 

 
 

其他參考項目： 

BOX 

 BARCODE 
 

功能：印製一維條碼，下方為所支援的條碼列表。 

 
 Code 128 (switching code subset automatically) 
 Code 128M (switching code subset manually) 
 EAN 128 (switching code subset automatically) 
 Interleaved 2 of 5 
 Interleaved 2 of 5 with check digit 
 Code 39 
 Code 39 with check digit 
 Code 93 
 EAN 13 
 EAN 13 with 2 digits add-on 
 EAN 13 with 5 digits add-on 
 EAN 8 
 EAN 8 with 2 digits add-on 
 EAN 8 with 5 digits add-on 
 Codabar 
 Postnet 
 UPC-A 
 UPC-A with 2 digits add-on 
 UPC-A with 5 digits add-on 
 UPC-E 
 UPC-E with 2 digits add-on 
 UPC-E with 5 digits add-on 
 MSI 
 PLESSEY  
 China POST 
 ITF14  
 EAN14 

語法： 

BARCODE X, Y, ”code type”, height, human readable, rotation, narrow, wide, “code” 
 

參     數   說          明 

X     明定條碼左上角 X座標 

Y     明定條碼左上角 X座標 

條碼類型    

128    Code 128，自動轉換 A、B、C的條碼格式。  

128M   Code 128，手動轉換 A、B、C的條碼格式。 

 

Control code A B C 

096 FNC3 FNC3 NONE 

097 FNC2 FNC2 NONE 

098 SHIFT SHIFT NONE 

099 CODE C CODE C NONE 

100 CODE B FNC4 CODE B 

101 FNC4 CODE A CODE A 

102 FNC1 FNC1 FNC1 

103 Start (CODE A) 

104 Start (CODE B) 

105 Start (CODE C) 

 

“!”是做條碼格式轉換使用，若未給定起始條碼的格式時，則以 code B做為起始條碼。 
 

EAN128   Code 128, 自動轉換 A，B，C條碼格式 

   25    Interleaved 2 of 5 

   25C    Interleaved 2 of 5含檢核碼 

   39    Code 39 

   39C    Code 39含檢核碼 

   93    Code 93  
   EAN13   EAN 13 
   EAN13+2  EAN 13 with 2 digits add-on 
   EAN13+5  EAN 13 with 5 digits add-on 
   EAN8   EAN 8  
   EAN8+2   EAN 8 with 2 digits add-on 
   EAN8+5   EAN 8 with 5 digits add-on 
   CODA   Codabar 
   POST   Postnet 
   UPCA   UPC-A 
   UPCA+2   UPC-A with 2 digits add-on 
   UPCA+5   UPC-A with 5 digits add-on 
   UPCE   UPC-E  
   UPCE+2   UPC-E with 2 digits add-on 
   UPCE+5   UPC-E with 5 digits add-on 
   CPOST   China post code 
   MSI    MSI code 
   MSIC   MSI code with check digital 

   PLESSEY  PLESSEY code 
   ITF14   ITF 14 code 
   EAN14   EAN 14 code 

 
height    條碼高度 (dot) 

human readable  0 : 無人眼可識別碼 

      1 : 有人眼可識別碼 

rotation    順時針旋轉條碼角度 direction 

   0    不旋轉 

   90    順時針旋轉 90度 direction 

   180    順時針旋轉 180度 direction 

   270    順時針旋轉 270度 direction 

narrow    窄條碼比例因子 (dot) 

wide     寬條碼比例因子 (dot) 

 

 narrow : wide 
1:1 

narrow : wide 
1:2 

narrow : wide 
1:3 

narrow : wide 
2:5 

narrow : wide 
3:7 

128 10x - - - - 

EAN128 10x - - - - 

25 - 10x 10x 5x - 

25C - 10x 10x 5x - 

39 - 10x 10x 5x - 

39C - 10x 10x 5x - 

93 - - 10x - - 

EAN13 8x - - - - 

EAN13+2 8x - - - - 

EAN13+5 8x - - - - 

EAN 8 8x - - - - 

EAN 8+2 8x - - - - 

EAN 8+5 8x - - - - 

CODA - 10x 10x 5x - 

POST 1x - - - - 

UPCA 8x - - - - 

UPCA+2 8x - - - - 

UPCA+5 8x - - - - 

UPCE 8x - - - - 

UPCE+2 8x - - - - 

UPCE+5 8x - - - - 

CPOST - - - - 1x 

MSI - - 10x - - 

MSIC   10x  - 

PLESSY - - 10x - - 

ITF14 - 10x 10x 5x - 

EAN14 - - - - - 

 

Barcode type Maximum bar code length 

128 - 

EAN128 - 

25 - 

25C - 

39 - 

39C - 

93 - 

EAN13 12 

EAN13+2 14 

EAN13+5 17 

EAN 8 7 

EAN 8+2 9 

EAN 8+5 12 

CODA - 

POST 5,9,11 

UPCA 11 

UPCA+2 13 

UPCA+5 16 

UPCE 6 

UPCE+2 8 

UPCE+5 11 

CPOST - 

MSI - 

MSIC  

PLESSY - 

ITF14 13 

EAN14 13 

  

範例： 

BARCODE 100,100,”39”,96,1,0,2,4,”1000” 
  BARCODE 10,10,”128M”,48,1,0,2,2,”!104!096ABCD!101EFGH” 
 

         (上述 CODE 128M 的範例為使用 CODE B 起始的條碼，其中!096(FNC3)與 ABCD均以

CODE B 方式編碼；!101為將原編碼方式由 CODE B 轉換為 CODE A，後續的 EFGH

即為使用 CODE A 方式編碼)

 BITMAP 
 

功能：繪製 BITMAP格式的圖形 (非 BMP 的圖形檔) 

    

語法：BITMAP X, Y, width, height, mode, bitmap data… 

 

參     數   說          明 

X     明定圖形左上角 X座標 

Y     明定圖形左上角 Y座標 

width    圖形的寬度，單位為 byte 

height    圖形的高度，單位為 dot 

mode    繪製圖形的方式 

 0        OVERWRITE 
 1        OR 
 2        XOR 

bitmap data   圖形資料  

 

範例： 
 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

ROW 
(Y- axis) 

L-Byte R-Byte 

Binary Hexadecimal Binary Hexadecimal 

0 00000000 00 00000000 00 

1 00000000 00 00000000 00 

2 00000000 00 00000000 00 

3 00000111 07 11111111 FF 

4 00000011 03 11111111 FF 

5 00010001 11 11111111 FF 

6 00011000 18 11111111 FF 

7 00011100 1C 01111111 7F 

8 00011110 1E 00111111 3F 

9 00011111 1F 00011111 1F 

A 00011111 1F 10001111 8F 

B 00011111 1F 11000111 C7 

C 00011111 1F 11100011 E3 

D 00011111 1F 11110111 F7 

E 00011111 1F 11111111 FF 

F 00011111 1F 11111111 FF 

 

範例: 

SIZE 4,2 
GAP 0,0 
CLS 
BITMAP 200,200,2,16,0,      � •-?  
PRINT 1,1 

 

Hexadecimal ASCII 

53 49 5A 45 20 34 2C 32 0D 0A 47 41 50 20 
30 2C 30 0D 0A 43 4C 53 0D 0A 42 49 54 4D 
41 50 20 32 30 30 2C 32 30 30 2C 32 2C 31 
36 2C 30 2C 00 00 00 00 00 00 07 FF 03 FF 
11 FF 18 FF 1C 7F 1E 3F 1F 1F 1F 8F 1F C7 
1F E3 1F E7 1F FF 1F FF 0D 0A 50 52 49 4E 
54 20 31 2C 31 0D 0A 

SIZE 4,2 
GAP 0,0 

CLS 
BITMAP 200,200,2,16,0,      
� •-?  

PRINT 1,1 
 

 
 

其他參考項目： 

PUTBMP, PUTPCX 

 BOX 
 

功能：繪製方框 

 

語法：BOX X_start, Y_start, X_end, Y_end, line thickness 

 

參     數   說          明 

X_start    方框左上角 X座標，單位：dot 

Y_start    方框左上角 Y座標，單位：dot 

X_end    方框右下角 X座標，單位：dot 

Y_end    方框右下角 X座標，單位：dot 

line thickness   方框線條粗細，單位：dot 

 

注意事項：200 DPI: 1 mm = 8 dots 

  300 DPI: 1 mm = 12 dots 

為保護電源供應器的正常運作，建議於列印 4 吋寬的線條時，最大高度不超過 12 

dots，否則將有可能損壞電源供應器及影響列印品質 

        另為保護印字頭的正常使用，對於桌上型機器與工業用機器單一點行最大的列印比率
分別為 20%及 30%。以 200dpi 機器為例，其單一點行為 864 dots，則桌上型機種
有效的列印範圍為 172 dots，工業型則為 259 dots。 

 

範例： 

SIZE 4,2.5 
GAP 0,0 
SPEED 6 
DENSITY 8 
DIRECTION 0 
CLS 
BOX 100,100,200,200,5 
PRINT 1,1 
 
 

其他參考項目： 

BAR 

 DMATRIX (V3.19) 
 

功能：繪製 DATAMATRIX二維條碼，目前僅支援 ECC200格式 

    

語法：DMATRIX x, y, width, height, [xm,row,col,], expression 

 

參     數   說          明 

x     條碼左上角 X座標，單位：dot 

y     條碼左上角 Y座標，單位：dot 

width    條碼寬度，單位：dot 

height    條碼高度，單位：dot 

xm      條碼密度(模組寬度)，單位：dot 

row                 條碼行碼，自 10 到 144（10，12，14，16，18，20，22，24，

26，32，36，40，44，48，52，64，72，80，88，96，104，120，132，144） 

col                  條碼欄碼，自 10到 144 

expression           條碼內容 

 

範例： 

SIZE 3,3 
GAP 0,0 
SPEED 4 
DENSITY 8 
DIRECTION 0 
REFERENCE 0,0 
OFFSET 0.00 
SET CUTTER OFF 
SET TEAR ON 
CLS 
DMATRIX 10,110,400,400,"DMATRIX EXAMPLE 1" 
DMATRIX 310,110,400,400,x6,"DMATRIX EXAMPLE 2" 
DMATRIX 10,310,400,400,x8,18,18,"DMATRIX EXAMPLE 3" 
PRINT 1,1 

 ERASE 
 

功能：清除 Image Buffer中指定大小的區域 

    

語法：ERASE X_start, Y_start, X_width, Y_height 

 

參     數   說          明 

X_start    欲清除區域的左上角 X座標，單位：dot 

Y_start    欲清除區域的左上角 Y座標，單位：dot 

X_width    欲清除區域的寬度，單位：dot 

Y_height    欲清除區域的高度，單位：dot 

 

範例： 

SIZE 4,2.5 
GAP 0,0 
SPEED 6 
DENSITY 8 
DIRECTION 0 
CLS 
BAR 100, 100, 300, 300 
ERASE 150,150,200,200 
PRINT 1,1 
 
 

其他參考項目： 

CLS 

 MAXICODE 
 

功能：繪製MAXICODE二維條碼 

    

語法：MAXICODE x, y, mode, [class, country, post, Lm,] “message” 

(1) Mode 2 or 3: 
MAXICODE x, y, mode, class, country, postal code, ”low priority message" 

若參數 country為 840，則參數 postal code的格式為 99999,9999 

若 country參數為其他值，則 postal code為最多 6個字母與文字符號 

   (2) Mode 4,5,6 
MAXICODE x, y, mode, [Lm], "message" 

 

*支援 AIM特殊格式，請見 MAXICODE規格書第 23頁 

             

注意事項： TTP-248M/246M/344M，TTP-245/343，TDP-245目前尚未支援Mode 6 格式 

 

參     數   說          明 

x     X-coordinate starting point條碼左上角 X座標，單位：dot 

y     Y-coordinate starting point條碼左上角 Y座標，單位：dot 

mode    2,3,4,5,6 

class    服務類型, 3碼數字 (僅提供mode 2,3使用)) 

country    國家代碼，3碼數字 (for mode 2,3) 

post     郵遞編號 (僅提供mode 2,3使用) 

      Mode 2: (USA) 格式為：5位數字,4位數字 

Mode 3: (Canada) 格式為：最多 6位字每與文字符號，需以雙引

號(“)括住。  

Lm 指定條碼內容長度(雙引號(”)可被忽略) , 1m138, 僅適用於

mode 4,5。 

message    條碼內容 

 

範例： 

SIZE 4,2 
GAP 0,0 
CLS 
SPEED 4 
DENSITY 8 
OFFSET 0.00 
REFERENCE 0,0 
SET CUTTER OFF 
SET TEAR ON 
 
REM *****Mode 2 For USA***** 
MAXICODE 110,100,2,300,840,06810,7317,"DEMO 2 FOR USA MAXICODE" 
TEXT 100,520,"3",0,2,2,"Mode 2 For USA" 
PRINT 1,1 
 
REM *****Mode 3 For Canada***** 
CLS 
MAXICODE 110,100,3,300,863,"107317","DEMO 3 FOR CANADA MAXICODE" 
TEXT 100,520,"3",0,2,2,"Mode 3 For CANADA" 
PRINT 1,1 
 
REM *****MODE4***** 

CLS 
MAXICODE 110,100,4,"DEMO 4 FOR MAXICODE" 
MAXICODE 600,100,4,19,DEMO 4 FOR MAXICODE 
TEXT 100,520,"3",0,2,2,"Mode 4 FOR MAXICODE" 
PRINT 1,1 
 
REM *****MODE 5***** 
CLS 
MAXICODE 110,100,5,"DEMO 5 FOR MAXICODE" 
MAXICODE 600,100,5,19,DEMO 5 FOR MAXICODE 
TEXT 100,520,"3",0,2,2,"DEMO 5 FOR MAXICODE" 
PRINT 1 

 PDF417 (參數 P,M,U 有問題) 
 

功能：繪製 PDF417二維條碼 

 

語法：PDF417 x, y, width, height, rotate, [option], expression 

 

參     數   說          明 

x     條碼左上角 X座標，單位：dot 

y     條碼左上角 Y座標，單位：dot 

width    條碼寬度，單位：dot 

height    條碼高度，單位：dot 

rotate    順時針旋轉角度 

 0:          不旋轉 

 90:         順時針旋轉 90度 

 180:        順時針旋轉 180度 

 270:        順時針旋轉 270度 

expression   條碼內容 

 
[option]     

P     資料壓縮方法 

      0: 自動生成 

      1: 二進制資料 

E     錯誤糾正等級 

      範圍：0~8 

M   定義可列印條碼區域的條碼生成位置 

0: 於可列印條碼區域的左上角開始向右向下繪製條碼 

      1: 於可列印條碼區域的正中央開始向外繪製條碼 

Ux,y,c    人眼可識別碼 

x: 生成人眼可識別碼的 X座標 

y: 生成人眼可識別碼的 Y座標 

c: 每行最多可顯示的字數 

W     模組寬度，單位：dot 

      範圍：2~9 

H     模組高度，單位：dot 

      範圍：4~99 

R     最大行數 

C     最大列數 

T     截斷結束區 

      0: 不截斷 

      1: 截斷 

Lm 指定條碼內容長度(不需使用雙引號(“)), 1m2048 

 
 
 

範例： 

SIZE 3,3 
GAP 0.12,0 
CLS 
SPEED 6 
DENSITY 8 

DIRECTION 1 
REFERENCE 0,0 
 
REM *****WITHOUR OPTIONS***** 
CLS 
PDF417 50,50,400,200,0,"Without Options" 
PRINT 1,1 
 
REM *****OPTION:E4***** 
CLS 
PDF417 50,50,400,200,0,E4,"Error correction level:4" 
PRINT 1,1 
 
REM *****OPTION:E4 W4***** 
CLS 
PDF417 50,50,600,600,0,E4,W4,"Error correction level:4 
module width 4 dots" 
PRINT 1,1 
 
 
REM *****OPTION:E4 W4 H4***** 
CLS 
PDF417 50,50,600,600,0,E4,W4,H4,"Error correction level:4 
module width 4 dots 
bar height 4 dots" 
PRINT 1,1 
 
 
REM *****OPTION:E4 W4 H4 R25***** 
CLS 
PDF417 50,50,600,600,0,E4,W4,H4,R25,"Error correction level:4 
Module Width 4 dots 
Bar Height 4 dots 
Maximum Number of Rows: 25 Rows 
" 
PRINT 1,1 
 
REM *****OPTION:E4 W4 H4 R40 C3***** 
CLS 
PDF417 50,50,600,600,0,E4,W4,H4,R40,C3,"Error correction level:4 
Module Width 4 dots 
Bar Height 4 dots 
Maximum Number of Rows: 40 Rows 
Maximum number of columns: 3 Cols 
" 
PRINT 1,1 
 
REM *****OPTION:E4 W4 H4 R40 C4 T0***** 
CLS 
PDF417 50,50,600,600,0,E4,W4,H4,R40,C4,T0,"Error correction level:4 
Module Width 4 dots 
Bar Height 4 dots 
Maximum Number of Rows: 40 Rows 
Maximum number of columns: 4 Cols 
Truncation:0 
" 



 

TAIWAN SEMICONDUCTOR CO., LTD.          37          Copyright 2001 All Rights Reserved. 
 

PRINT 1,1 
 
REM *****OPTION:E4 W4 H4 R40 C4 T1***** 
CLS 
PDF417 50,50,900,900,0,E4,W4,H4,R40,C4,T1,"Error correction level:4 
Module Width 4 dots 
Bar Height 4 dots 
Maximum Number of Rows:5 Rows 
Maximum number of columns:90 Cols 
Truncation:1 
" 
PRINT 1,1 
 
REM *****OPTION:E4 W4 H4 R40 C4 T0 L169***** 
CLS 
PDF417 50,50,900,900,0,E4,W4,H4,R40,C4,T0,L169,Error correction level:4 
Module Width 4 dots 
Bar Height 4 dots 
Maximum Number of Rows: 40 Rows 
Maximum number of columns: 4 Cols 
Truncation:0 
Expression length:167 
PRINT 1,1 
 
REM *****OPTION:E4 W4 H4 R40 C4 T1 L169***** 
CLS 
PDF417 50,50,900,900,0,E4,W4,H4,R40,C4,T1,L169,Error correction level:4 
Module Width 4 dots 
Bar Height 4 dots 
Maximum Number of Rows: 40 Rows 
Maximum number of columns: 4 Cols 
Truncation:1 
Expression length:169 
PRINT 1,1 
 
REM *****OPTION:P0 E4 W4 H4 R40 C4 T1 L169***** 
CLS 
PDF417 50,50,900,900,0,P0,E4,W4,H4,R40,C4,T1,L169,Error correction level:4 
Module Width 4 dots 
Bar Height 4 dots 
Maximum Number of Rows: 40 Rows 
Maximum number of columns: 4 Cols 
Truncation:1 
Expression length:169 
PRINT 1,1 
 
REM *****OPTION:P0 E4 M0 W6 H6 R60 C4 T0 L283***** 
SIZE 3,2 
CLS 
PDF417 50,50,900,600,0,P0,E4,M0,W6,H6,R60,C4,T0,L283,Data compression 
method: P0 
Error correction level: E4 
Center pattern in barcode area: M0 
Human Readable: No 
Module Width 6 dots: W6 
Bar Height 6 dots: H6 

Maximum Number of Rows: 60 Rows: R60 
Maximum number of columns: 4 Cols: C4 
Truncation:0: T0 
Expression length:283: L283 
PRINT 1,1 
 
REM *****OPTION:P1 E4 M1 U100,500,10 W4 H4 R60 C4 T1 L297***** 
CLS 
PDF417 50,50,900,600,0,P1,E4,M1,U100,500,10,W6,H6,R60,C4,T1,L297,Data 
compression method: P1 
Error correction level: E4 
Center pattern in barcode area: M1 
Human Readable: Yes: U100,300,10 
Module Width 6 dots: W6 
Bar Height 6 dots: H6 
Maximum Number of Rows: 60 Rows: R60 
Maximum number of columns: 4 Cols: C4 
Truncation:1: T1 
Expression length:297: L297 
PRINT 1,1 

 PUTBMP 
 

功能：列印 BMP格式圖檔 

    

語法：PUTBMP X, Y, ”filename” 

 

參     數   說          明 

X     BMP圖形左上角 X座標 

Y     BMP圖形左上角 Y座標 

filename    已載至印表機的 BMP圖檔 

 

注意事項：建議使用僅有黑白二種顏色的 BMP圖檔，最多可支援 256色的 BMP圖檔。目
前 僅 下 列 機 種 支 援 此 項 指 令 ： TTP-2410M/248M/246M/344M ，
TTP-245/245G/343及 TDP-245。 

 

 範例：^Z表示<Ctrl>+<z>或者是<F6>鍵 

   
   
   
   
   
   
   
   
   
   
   
   
   
   
   
   
   
   
 
 
 
   

其他參考項目： 

DOWNLOAD, BITMAP, PUTPCX 

 PUTPCX 
 

功能：列印 PCX格式圖檔 

   語法：PUTPCX X, Y, ”filename” 

 

參     數   說          明 

X     PCX圖形左上角 X座標 

Y     PCX圖形左上角 Y座標 

filename    已載至印表機的 PCX圖檔 

 

注意事項：建議使用僅有黑白二種顏色的 PCX圖檔，最多可支援 256色的 PCX圖檔。目
前 僅 下 列 機 種 支 援 此 項 指 令 ： TTP-2410M/248M/246M/344M ，
TTP-245/245G/343及 TDP-245。 

 

範例：^Z表示<Ctrl>+<z>或者是<F6>鍵 

   
         
         
         
         
         
         
   
   
   
   
   
   
   
   
   
   
   
   
         
   
   

 
 
其他參考項目： 

DOWNLOAD, BITMAP, PUTPCX 

 QRCODE 
 

功能：繪製 QRCODE二維條碼 
   

語法： 

QRCODE X, Y, ECC Level, cell width, mode, rotation, [model, mask,]"Data string” 

參     數   說          明 

X     QRCODE條碼左上角 X座標 

Y     QRCODE條碼左上角 Y座標 

ECC level   錯誤糾正能力等級 

L    7% 
  M    15% 
  Q    25% 
  H    30% 
cell width    1~10 

mode    自動生成編碼/手動生成編碼 

A    Auto 
  M    Manual 

rotation    順時針旋轉角度 

0    不旋轉 

  90    順時針旋轉 90度 

  180    順時針旋轉 180度 

  270    順時針旋轉 270度 

model    條碼生成樣式 

1     (預設), 原始版本 

2    擴大版本 

mask    範圍：0~8，預設 7 

Data string   條碼資料內容 

 

可用編碼字元集： 

1). 數值資料：數字 0~9 

2). 文字數字資料：數字 0~9；大寫字母 A-Z; 其他：space，$%*+-./:  

3). 8-bit 二進制資料(JIS 8-bit 字元表(拉丁語及假名) 與 JIS X 0201相符) 

4).日文漢字字元 (Shift JIS values 8140HEX –9FFCHEX and E040HEX –EAA4 HEX. 

These are values shifted from those of JIS X 0208. Refer to JIS X 0208 Annex 1 
Shift Coded Representation for detail.). 

  

條碼的最大資料長度： 

                     Model 1 (Version 14-L)   Model 2 (Version 40-L) 

          1). 數值資料：        1,167 characters        7,089 characters 

2). 文字數字資料：   707 characters          4,296 characters 

3). 8-bit 二進制資料：  486 characters          2,953 characters 

4). 日文漢字資料:      299 characters          1,817 characters 

 
於手動生成編碼下： 

1. 若條碼內容的第一個字元為”A”，則後續的資料為”文字數字”型態。 

2. 若條碼內容的第一個字元為”N”，則後續的資料為”數字”型態。 

3. 若條碼內容的第一個字元為”B”，則後續的 4 位數字表示緊接二進位資料的長度(單

位為 byte)，且後續的資料為”二進制資料”型態。 

4. 若條碼內容的第一個字元為”K”，則後續的資料為”日文漢字”型態。 

5. “!”則是用來轉換資料的格式，”N”、”A”、”B”、”K”等資料型態可透過”!”的轉換組成一

組條碼內容。 

    

 範例：   手動生成 QRCODE的範例: 

QRCODE 100,10,L,7,M,0,M1,S1,"ATHE FIRMWARE HAS BEEN UPDATED" 
(A：文字數字混合型態的資料)  

QRCODE 100,10,M,7,M,0,M1,S2,"N123456" 
(N：數值型態的資料)  

QRCODE 100,10,Q,7,M,0,M1,S3,"N123456!ATHE FIRMWARE HAS BEEN 
UPDATED" 
(N：數值型態的資料；!：資料型態轉換用字元；A；文字數字混合型態的資料) 

QRCODE 100,10,H,7,M,0,M1,S3,"B0012Product name" 

(B：二進制資料；0012：12 bytes ) 

QRCODE 100,10,M,7,M,0,M1,S3,"K印表機" 

(K：日文漢字資料) 

 
自動生成 ORCODE的範例: 

QRCODE 100,10,M,7,A,0,"THE FIRMWARE HAS BEEN UPDATED" 
 

(1) 自動生成 ORCODE的範例： 

a. 一般資料字串 

SIZE 4,2.5 
GAP 0.12,0 
CLS 
QRCODE 10,10,H,4,A,0,"ABCabc123" 
QRCODE 160,160,H,4,A,0,"123ABCabc" 

QRCODE 310,310,H,4,A,0,"印表機 ABCabc123" 

PRINT 1,1 

b. 資料字串中包含<Enter>(即 0Dh, 0Ah) 

SIZE 4,2.5 
GAP 0.12,0 
CLS 
QRCODE 10,10,H,4,A,0,"ABC<Enter> 
abc<Enter> 
123" 
QRCODE 160,160,H,4,A,0,"123<Enter> 
ABC<Enter> 
abc" 

QRCODE 310,310,H,4,A,0,"印表機<Enter> 

ABC<Enter> 
abc<Enter> 
123" 

PRINT 1,1 

c. 資料串接 (必須在 DOENLOAD…EOP模式下) 

DOWNLOAD “DEMO.BAS” 
SIZE 4,2.5 
CAP 0.12,0 
CLS 
QRCODE 10,10,H,4,A,0,"ABCabc123"+STR$(1234) 
QRCODE 160,160,H,4,A,0,"123ABCabc"+"1234" 

QRCODE 310,310,H,4,A,0,"印表機 ABCabc123"+"1234"+"abcd" 

PRINT 1,1 
EOP 

DEMO 

d. 資料內容含有雙引號(“)時，請用\[“]取代 

SIZE 4,2.5 
CAP 0.12,0 
CLS 
QRCODE 10,10,H,4,A,0,"ABC\["]abc\["]123" 
QRCODE 160,160,H,4,A,0,"123\["]ABC\["]abc" 

QRCODE 310,310,H,4,A,0,"\["]印表機\["]ABCabc123" 

PRINT 1,1 

(2) Manual mode 

a. 一般資料字串： 

SIZE 4,2.5 
CAP 0.12,0 
CLS 
QRCODE 10,10,H,4,M,0,"AABC!B0003abc!N123" 
QRCODE 160,160,H,4,M,0,"N123!AABC!B0003abc" 

QRCODE 310,310,H,4,M,0,"K印表機!AABC!B0006abc123" 

PRINT 1,1 

b. 資料字串中包含<Enter>(即 0Dh, 0Ah), <Enter> 則為 8-bit byte data 

SIZE 4,2.5 
CAP 0.12,0 
CLS 
QRCODE 10,10,H,4,M,0,"AABC!B0007<Enter> 

abc<Enter> 
!N123" 
QRCODE 160,160,H,4,M,0,"N123!B0002<Enter> 
!AABC!B0005<Enter> 
abc" 

QRCODE 310,310,H,4,M,0,"K印表機!B0002<Enter> 

!AABC!B0010<Enter> 
abc<Enter> 
123" 

PRINT 1,1 

c. 資料串接 (必須在 DOENLOAD…EOP模式下) 

DOWNLOAD “A.BAS” 
SIZE 4,2.5 
CAP 0.12,0 
CLS 

QRCODE 10,10,H,4,M,0,"AABC!B0006abc123!N"+STR$(1234) 
QRCODE 160,160,H,4,M,0,"N123!AABC!B0007abc"+"1234" 

QRCODE 310,310,H,4,M,0,"K印表機!AABC!B0014abc123"+"1234"+"abcd" 

      PRINT 1,1 
      EOP 
      A 

d. 資料內容含有雙引號(“)時，請用\[“]取代 

SIZE 4,2.5 
CAP 0.12,0 
CLS 

QRCODE 10,10,H,4,M,0,"AABC!B0005\["]abc\["]!N123" 
QRCODE 160,160,H,4,M,0,"N123!B0001\["]!AABC!B0004\["]abc" 

QRCODE 310,310,H,4,M,0,"B0001\["]!K印表機!B0010\["]ABCabc123" 

PRINT 1,1 

 REVERSE 
 

功能：將 Image Buffer中指定的區域做反相的動作 

    

語法：REVERSE X_start, Y_start, X_width, Y_height 

 

參     數   說          明 

X_start    欲清除區域的左上角 X座標，單位：dot 

Y_start    欲清除區域的左上角 Y座標，單位：dot 

X_width    欲清除區域的寬度，單位：dot 

Y_height    欲清除區域的高度，單位：dot 

 

注意事項：200 DPI: 1 mm = 8 dots 

300 DPI: 1 mm = 12 dots 

為保護電源供應器的正常運作，建議於列印 4吋寬經反相後得到的線條時，最大高度
不超過 12 dots，否則將有可能損壞電源供應器及影響列印品質。另為保護印字頭的
正常使用，對於桌上型機器與工業用機器單一點行最大的列印比率分別為 20%及
30%。以 200dpi機器為例，其單一點行為 864 dots，則桌上型機種有效的列印範圍
為 172 dots，工業型則為 259 dots。 

 

 範例： 

SIZE 4,2.5 
GAP 0,0 
SPEED 6 
DENSITY 8 
DIRECTION 0 
CLS 
TEXT 100,100,"3",0,1,1,"REVERSE" 
REVERSE 90,90,128,40 
PRINT 1,1 

 TEXT 
 

功能：列印文字 

    

語法：TEXT X, Y, ”font”, rotation, x-multiplication, y-multiplication, “content” 

 

參     數   說          明 

X     文字方塊左上角 X座標 

Y     文字方塊左上角 Y座標 

font     字型名稱 

  1     8 x 12 fixed pitch dot font 
  2     12 x 20 fixed pitch dot font 
  3     16 x 24 fixed pitch dot font 
  4     24 x 32 fixed pitch dot font 
  5     32 x 48 dot fixed pitch font 
  6     14 x 19 dot fixed pitch font OCR-B 

7 14 x25 dot fixed pitch font OCR-A 
8 21 x 27 dot fixed pitch font OCR-B 

  ROMAN.TTF   ROMAN True type font 
 

Rotation     順時針旋轉角度  

  0     不旋轉 

  90     順時針旋轉 90度 

  180     順時針旋轉 180度 

270 順時針旋轉 270度 

 

X-multiplication:  水平放大值，最大可放大至 10倍  

有效係數：1~10 

使用仿真字型時，此參數可被忽略 

Y-multiplication:  垂直放大值，最大可放大至 10倍 

有效係數：1~10 

使用仿真字型時，此參數決定 TTF字的高度，其中 

1 point=1/72 inch 
 

注意事項：若文字內容包括雙引號(“)時，則請用\[“] 取代 

          字型 1及 2, 每個字元的間隙為水平放大值*2 dots 

          字型 3及 4, 每個字元的間隙為水平放大值-1 dots 

          字型 5, 每個字元的間隙為水平放大值*5 dots 

 

 範例 1： 

SIZE 3,2 
GAP 0,0 
CLS 
TEXT 100,100,"5",0,1,1,"\["]DEMO FOR TEXT\["]" 
TEXT 100,200,"ROMAN.TTF",0,1,20,"\["]True Type Font Test Print\["]" 
PRINT 1,1 

 範例2： 

DOWNLOAD F,"test.BAS" 
SPEED 4 
DENSITY 8 
SET CUTTER OFF 
SET PEEL OFF 

DIRECTION 0 
SIZE 1.000,0.800 
GAP 0.12,0.00 
OFFSET 0.000 
REFERENCE 0,0 
:START 
CLS 
INPUT "Description",A$ 
Again: 
F$="" 
INPUT "Font:",1,F$ 
IF ASC(F$)>53 THEN GOTO Again 
IF ASC(F$)<49 THEN GOTO Again 
X_Again: 
INPUT "X_multiple:",1,XM 
IF XM>8 THEN GOTO X_Again 
IF XM<1 THEN GOTO X_Again 
Y_Again: 
INPUT "Y_multiple:",1,YM 
IF YM>8 THEN GOTO Y_Again 
IF YM<1 THEN GOTO Y_Again 
 
Z=LEN(A$) 
WWidth=Z*XM*8 
IF F$="1" THEN 
   REM *****When Font1, the space between every single word is XM*2 dots***** 
   WWidth=Z*(XM*8) 
   SWidth=Z*(XM*2) 
   X=101-((WWidth+SWidth)/2) 
ELSEIF F$="2" THEN 
   REM *****When Font2, the space between every single word is XM*2 dots***** 
   WWidth=Z*(XM*12) 
   SWidth=Z*(XM*2) 
   X=101-((WWidth+SWidth)/2) 
ELSEIF F$="3" THEN 
   REM *****When Font3, the space between every single word is XM-1 dots***** 
   WWidth=Z*(XM*16) 
   SWidth=Z*(XM-1) 
   X=101-((WWidth+SWidth)/2) 
ELSEIF F$="4" THEN 
   REM *****When Font4, the space between every single word is XM-1 dots***** 
   WWidth=Z*(XM*24) 
   SWidth=Z*(XM-1) 
   X=101-((WWidth+SWidth)/2) 
ELSEIF F$="5" THEN 
   REM *****When Font5, the space between every single word is XM*4 dots***** 
   WWidth=Z*(XM*32) 
   SWidth=Z*(XM*4) 
   X=101-((WWidth+SWidth)/2) 
ENDIF 
TEXT X,116,F$,0,XM,YM,A$ 
PRINT 1,1 
GOTO START 
EOP 

Status Polling Commands (RS-232) 
 

 <ESC>!? 
 

 功能：透過串列埠送出指令用來獲得印表機目前狀態，其中<ESC>跳脫符號表示 ASCII 27 

(Hex1B)，此項指令可隨時被送出，即使是在印表機發生錯誤狀態時。 

 

位元      狀     態 

  0      Ready 

01          印字頭開啟 Printer carriage open 

12       卡紙 

23       缺紙 

34       缺碳帶 

45       暫停  

56       列印中 Reserved 

67       外蓋開啟(選購) 

7       環境溫度超過範圍 (選購) 

 

位元組 16進制 印表機狀態 

00000000 00 待機中 

00000001 01 印字頭開啟 

00000010 02 卡紙 

00000011 03 卡紙並開啟印字頭 

00000100 04 缺紙 

00000101 05 缺紙並開啟印字頭 

00001000 08 缺碳帶 

00001001 09 缺碳帶並開啟印字頭 

00001010 0A 缺碳帶並卡紙 

00001011 0B 缺碳帶並卡紙及開啟印字頭 

00001100 0C 缺碳帶並同時缺紙 

00001101 0D 缺碳帶、缺紙並開啟印字頭 

00010000 10 暫停 

00100000 20 列印中 

 

語法：<ESC>!? 

 

其他參考項目： 

<ESC>!R 

 <ESC>!R 
 

功能：透過串列埠送出指令使印表機重新開機，其中<ESC>跳脫符號表示 ASCII 27 (Hex1B)，此

項指令可隨時被送出，重新開機後 DRAM中的資料將被清除。 

 

語法：<ESC>!R 

 

其他參考項目： 

<ESC>!? 

 ~!@ (New item) (Modified by V3.16) 
 

功能：透過串列埠詢問印表機已列印的哩程數(單位：公尺)。哩程數的整數部份將被回傳，小數

部份則被忽略。回傳的型態為 ASCII字元，結尾部份為 0x0D。 

    

語法：~!@ 

 

   範例： 

~!@ 

 ~!A (New item) (Modified by V3.16) 
 

功能：透過串列埠回傳印表機記憶體剩餘空間，回傳時將分別列出 DRAM、FLASH、EXT. FLASH

等記憶體的剩餘空間，以十進制方式回傳，並以 0x0d做結束。 

 

語法：~!A 

 

範例： 

~!A 
 
其他參考項目： 

FILES 

 ~!C (New item) (Modified by V3.16) 
 

功能：透過串列埠回傳印表機是否有安裝 RTC。 

 

語法：~!C 

 

回覆值              說          明 

0                   未安裝 RTC 

1                   已安裝 RTC 

 

範例： 

~!C 
 

其他參考項目： 

YEAR, MONTH, DATE, WEEK, HOUR, MINUTE, SECOND, @YEAR, @MONTH, 
@DATE, @DAY, @HOUR, @MINUTE, @SECOND 

 ~!D (New item by V3.19) 
 

功能：進入“偵錯模式” 

    

語法：~!D 

    

   範例： 

~!D 

 ~!F 
 

功能：透過串列埠查詢回傳印表機內存在的檔案名稱或是字型名稱。檔名以 ASCII 字元回傳，

檔名與檔名之間以 0x0D間隔，回傳字串的結束字元為 0x0A。 

    

語法：~!F 

 

範例： 

~!F 
 

其他參考項目： 

FILES 

 ~!I 
 

功能：查詢印表機內的 CODE PAGE及 COUNTRY的設定值 

      經 RS-232回傳的格式為： 

      例：8 bit : 437,001 

7 bit : USA,001 

         有關印表機所支援的 CODE PAGE及 COUNTRY，請分別參閱 CODEPAE 

及 COUNTRY指令 

    

語法：~!I 

 

範例： 

~!I 
 

其他參考項目：: 

COUNTRY, CODEPAGE 

 ~!T 
 

功能：查詢印表機的型號，經 RS-232以 ASCII字元回傳 

  

印表機種類 回傳字串 

TTP-2410M TTP2410M 

TTP-248M TTP248M 

TTP-246M TTP246M 

TTP-246G TTP246G 

TTP-344M TTP344M 

TTP-245 TTP245 

TDP-245 TDP245 

TTP-245G TTP245G 

TDP-245G TDP245G 

TTP-343 TTP343 

    

語法：~!T 

 

範例： 

~!T 
 

其他參考項目：: 

~!I, ~!F 

Message Translation Protocols 
 ~# 

 

功能：將提示字串顯示於KP-200 LCD螢幕上，其中起始識別符號~#會經由印表機傳送至KP-200

鍵盤，並以~&結束。 

@0置於~&後方用來決定提示字串置於 LCD的上行 

@1置於~&後方用來決定提示字串置於 LCD的下行 

當@0或@1被忽略時，則提示字元置於上行，資料輸入則會顯示於下行 

 

 語法：~#Prompt~&[@0] 

~#Prompt~&[@1] 
 

 範例： 

DOWNLOAD "A.BAS" 
OUT "~#KP-200~&@0" 
OUT "~#Testing~&@1" 
EOP 

 
其他參考項目： 

INPUT, OUT 

mailto:~#Prompt~&@0
mailto:~#Prompt~&@0

Commands for Windows Driver 
 !B 
 

功能：用來將 BITMAP圖形儲存於印表機記憶體中，其中 nnn表示 BITMAP 圖檔的

資料內容。 

       

      注意事項：此指令僅提供予 TTP-243/342 系列，243M/342M 系列及
643/643plus系列使用。 

 

語法：!Bnnn 

 

參     數   說          明 

nnn     PC傳至印表機的圖檔資料位元數，以 3位數字表示 

 

範例： 

!B100 
 

其他參考項目： 

BITMAP 



 

TAIWAN SEMICONDUCTOR CO., LTD.          58          Copyright 2001 All Rights Reserved. 
 

 

 !J 
 

功能：定義 BITMAP圖形的列印位置 (Y座標) 

      注意事項：此指令僅提供予 TTP-243/342 系列，243M/342M 系列及
643/643plus系列使用。 

 

 語法： 

!Jnnnn 
 

 參     數   說          明 

nnnn    圖形輸出的 Y座標，以 4位數字表示 

 

 範例： 

!J0100 
 

其他參考項目： 

FEED 

 !N (add) 
 

功能：定義標籤的列印張數 

      注意事項：此指令僅提供予 TTP-243/342 系列，243M/342M 系列及
643/643plus系列使用。 

 

 語法： 

!Nnnn 
 

 參     數   說          明 

nnn       每張標籤的列印張數 

 

 範例： 

!N001 

File Management Commands 
 DOWNLOAD 
 

功能：”DOWNLOAD” 為一檔案內容的檔頭(即檔案首行)，定義檔案可被儲存於印表機的記憶體

中。下載至印表機的檔案可區分為兩種：程式檔及資料檔 (包括文字資料檔、PCX 檔形

檔、BMP圖形檔及 BITMAP字型檔) 

有關不同檔案類型的 DOWNLOAD 語法詳述如下： 

   

可儲存最大檔案數量： 

印表機型號 DRAM FLASH EXT. FLASH 

TTP-246M/344M 50 256 256 

TTP-245/343/TDP-245 50 64 256 

 

        當所載至印表機的程式檔名為”AUTO.BAS”時，該程式可於開機時自動執行。消取自動執

行”AUTO.BAS”的方式如下： 

   
  TTP-245/TTP-343/TDP-245 (Since V3.19) 

  按住 FEED鍵並開啟印表機電源，此時 LED燈號開始依序閃爍： 

橘燈紅燈(閃 5次) 橘燈(閃 5次) 綠燈(閃 5次) 綠燈(不閃) 

當燈號維持在綠燈不閃爍時鬆開 FEED 鍵，則”AUTO.BAS”將不執行，且印表機維持在”

待機”狀態。 

 

       TTP-246M/TTP-344M 系列 

     按住 FEED 鍵並開啟印表機電源，待 ERROR LED 燈號亮起時鬆開 FEED 鍵，此時即可

忽略”AUTO.BAS”，且印表機維持在”待機”狀態。 

 

 語法： 

1. 下載程式檔： 

DOWNLOAD [n,]“FILENAME.BAS” 

  參     數   說          明 

n 指定儲存檔案的位置 

 不指定時: 檔案被儲存於 DRAM. 

F：檔案被儲存於主機板的 FLASH. 

 …… 
  

E：檔案被儲存於外接式的 FLASH. 

FILENAME.BAS  儲存於印表機記憶體的檔案名稱 

 

注意事項： 

         (1). 大小寫將表示不同的檔案名稱 

         (2). 程式檔案的副檔名必須為”.BAS” 

        (3). 檔案名稱的格式必為 8.3(即主檔名長度為 8個字元，副檔名為 3個字元) 

        (4). 若不指定記憶體的位置，則檔案一律載至 DRAM，存於 DRAM的檔案會
因電源關閉而消失 

  

2. 下載資料檔案： 

DOWNLOAD [n,]“FILENAME”, DATA SIZE, DATA CONTENT…where 
 

  參     數   說          明 

n 指定儲存檔案的位置 

 不指定時: 檔案被儲存於 DRAM. 

F：檔案被儲存於主機板的 FLASH. 

 …… 
  

E：檔案被儲存於外接式的 FLASH. 

FILENAME  The name of data file that will remain resident in the 
printer memory. It is case sensitive. 

DATA SIZE 不含檔頭的實際檔案大小，以位元數計算 

 

注意事項： 

         (1). 行與行之間的資料以 CR(0x0D)及 LF(0x0A)做區隔 

       (2). 若不指定記憶體的位置，則檔案一律載至 DRAM，存於 DRAM的檔案
會因電源關閉而消失 

 

 範例： 

        下方程式範例會將檔案載至印表機的 DRAM 

  DOWNLOAD “EXAMPLE.BAS” 
  SIZE 4,4 
  GAP 0,0 
  DENSITY 8 
  SPEED 6 
  DIRECTION 0 
  REFERENCE 0,0 
  SET CUTTER OFF 
  SET PEEL OFF 
  CLS 
  TEXT 100,100,”3”,0,1,1,”EXAMPLE PROGRAM” 
  PRINT 1 
  EOP 
 

        注意事項：當撰寫一個可被下載至印表機的程式時，檔頭”DOWNLOAD”必須置於檔案的
首行，”EOP”必須置於程式的最後一行。 

                  執行程式的方式有二：一為呼叫主檔名(不含 BAS)執行，另一為使用 RUN 指
令執行程式。 

 

範例: 

1. 呼叫主檔名執行： 

C:\>COPY CON LPT1<ENTER> 
EXAMPLE<ENTER> 
<CTRL><Z> 
C:\> 
 

2. 使用 RUN指令執行程式： 

C:\>COPY CON LPT1<ENTER> 
RUN “EXAMPLE.BAS”<ENTER> 
<CTRL><Z> 
C:\> 

 

下方為下載資料檔案的範例. 

DOWNLOAD “DATA”,20,COMPUTER<Enter> 
2001<Enter> 
21<Enter> 

 

        注意事項：上方範例中<ENTER>表示按下鍵盤上的”ENTER”鍵。 



 

TAIWAN SEMICONDUCTOR CO., LTD.          62          Copyright 2001 All Rights Reserved. 
 

 
其他參考項目： 

EOP, RUN, PUTBMP, PUTPCX, INPUT 

 EOP 
 

功能：做為載入程式檔的結尾。當使用 BASIC語法時，DOWNLOAD “FILENAME.”必須置於檔

案首行，EOP必須置於檔案結尾。 

 

語法： 

EOP 
 

範例： 

DOWNLOAD “DEMO.BAS” 
SIZE 4,4  
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 100,100,”3”,0,1,1,”DEMO PROGRAM” 
PRINT 1 
EOP 

 
其他參考項目： 

DOWNLOAD, EOP, INPUT 

 FILES 
 

功能：此指令會將已載入印表機記憶體的檔案名稱印出或透過 RS-232傳回(包含 

DRAM、FLASH、EXT. FLASH)，並印出總記憶體大小及可用記憶體大小 

 

注意事項： 

TTP-243(E)/342/243M/342M系列為透過 RS-232將檔名回傳 

TTP-245/343/TDP245/TTP-246M/344M則透過列印方式將檔名印出 

 

語法： 

FILES 
 

範例： 

     下列步驟為在 DOS 模式下透過並列埠或串列埠將印表機內的檔案名稱及記憶體剩餘空間

印出或回傳 

 

串列埠： 

C:\>MODE COM1 96,N,8,1<ENTER> 
C:\>COPY CON COM1<ENTER> 

FILES<ENTER> 
<CTRL><Z><ENTER> 

C:\> 
 

並列埠： 

C:\>COPY CON LPT1<ENTER> 
FILES<ENTER> 
<CTRL><Z><ENTER> 

 

注意事項：上方範例中<ENTER>表示按下鍵盤上的”ENTER”鍵，<CTRL>表示按下鍵盤
上的”Ctrl”鍵，<Z>表示按下鍵盤上的”Z”鍵。 

 
 
其他參考項目： 

~!F, KILL 

 KILL 
 

功能：刪除記憶體內的檔案，萬用字元”*”可將 DRAM或 FLASH內的所有資料刪除。 

 

語法： 

KILL [n],”FILENAME” 
 

參     數   說          明 

n  指定要刪除檔除的位置 

 未指定時：刪除 DRAM中的檔案. 

F：刪除主機板上 FLASH中的檔案. 

 …… 
  

E：刪除外接式記憶卡內的資料. 

注意事項：當未指定刪除檔案的位置時，將刪除 DRAM中的檔案 

 

語法範例： 

1. KILL “FILENAME” 
2. KILL “*.PCX” 
3. KILL “*” 
4. KILL F,“FILENAME” 
5. KILL E,“*.PCX” 

 

 範例： 

    使用者可使用”SELFTEST”指令、”FILES”指令或”~!F”獲得印表機記憶體內的檔案名稱，

下列步驟可透過並列埠刪去指定位置的檔案。 

 
C:\>COPY CON LPT1<ENTER> 

FILES<ENTER> 
<CTRL><Z><ENTER> 

C:\>COPY CON LPT1<ENTER> 
KILL “DEMO.BAS” <ENTER> 
<CTRL><Z><ENTER> 

C:\>COPY CON LPT1<ENTER> 
FILES<ENTER> 
<CTRL><Z><ENTER> 

 

注意事項：上方範例中<ENTER>表示按下鍵盤上的”ENTER”鍵，<CTRL>表示按下鍵盤
上的”Ctrl”鍵，<Z>表示按下鍵盤上的”Z”鍵。 

 
其他參考項目： 

~!F, FILES 

 MOVE 
 

功能：此指令可將原下載至 DRAM的檔案移至主機板的 FLASH內 

 

語法： 

MOVE 
 

其他參考項目： 

DOWNLOAD, EOP 



 

TAIWAN SEMICONDUCTOR CO., LTD.          67          Copyright 2001 All Rights Reserved. 
 

 

 RUN 
 

功能：用來執行已存於記憶體中的程式。 

 

語法： 

RUN “FILENAME.BAS” 
 

範例： 

C:\>COPY CON LPT1<ENTER> 
RUN “DEMO.BAS”<ENTER> 
<CTRL><Z><ENTER> 
C:\> 
 

注意事項：上方範例中<ENTER>表示按下鍵盤上的”ENTER”鍵，<CTRL>表示按下鍵盤
上的”Ctrl”鍵，<Z>表示按下鍵盤上的”Z”鍵。 

 
其他參考項目： 

DOWNLOAD, EOP 

BASIC Commands and Functions 
 ABS( ) 
 

    功能：此函數回傳一整數或浮點數(小數)或一變數的絕對值。 

 

語法： 

ABS (-100) 
ABS (-99.99) 
ABS (VARIABLE) 
 

範例： 

DOWNLOAD “TEST.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
A=ABS(-100) 
B=ABS(-50.98) 
C=-99.99 
TEXT 100,100,”3”,0,1,1,STR$(A) 
TEXT 100,150,”3”,0,1,1,STR$(B) 
TEXT 100,200,”3”,0,1,1,STR$(ABS(C)) 
PRINT 1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP 

 ASC( ) 
 

  功能：回傳一字元的 ASCII碼。 

 

語法： 

ASC (“A”) 
 

範例： 

DOWNLOAD “TEST.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
CODE1=ASC(“A”) 
TEXT 100,100,”3”,0,1,1,STR$(CODE1) 
PRINT 1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, STR$() 

 CHR$( ) 
 

功能：回傳一指定 ASCII碼所代表的字元。 

 

語法： 

CHR$(n) 
 

參     數   說         明 

n     ASCII碼. 

 

 範例： 

DOWNLOAD “TEST.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
A=65 
WORD$=CHR$(A) 
TEXT 100,100,”3”,0,1,1,WORD$ 
PRINT 1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, STR$(), ASC$() 

 END 
 

    功能：結束程式。 

 

 語法： 

END 
 

 範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 4,2 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 200,60,"4",0,1,1,"END COMMAND TEST" 
X=300 
Y=200 
X1=500 
Y1=400 
GOSUB DR_LINE 
PRINT 1 
END 
 
:DR_LINE 
FOR I=1 TO 100 STEP 10 
BOX X+I,Y+I,X1-I,Y1-I,5 
NEXT 
RETURN 
EOP 
DEMO 
 

其他參考項目： 

DOWNLOAD, EOP 

 EOF( ) (Modified by V3.19) 
 

功能：當開啟一已下載的檔案時，判斷檔案是否已到結尾處。 

 

語法： 

EOF (File Handle) 
 

參     數   說          明 

File handle    0或 1  

 

回   傳   值   說          明 

None-zero   已到達檔案尾端 

0     尚未到達檔案尾端 

 

範例： 

DOWNLOAD "DATA",16,COMPUTER 
2000 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.0,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
OPEN "DATA",0 
SEEK 0,0 
Y=110 
TEXT 10,10,"3",0,1,1,"*****EOF TEST*****" 
:A 
Temp$="" 
READ 0,ITEM$,P 
TEXT 10,Y,"2",0,1,1,ITEM$+"$"+STR$(P)+"[EOF(0)="+STR$(EOF(0))+"]" 
BARCODE 10,Y+25,"39",40,1,0,2,4,"PRICE-"+STR$(P) 
Y=Y+100 
IF EOF(0)=0 THEN GOTO A 
PRINT 1 
EOP 
DEMO 
 

其他參考項目： 

DOWNLOAD, EOP, OPEN, READ, SEEK 

 OPEN (Modified by V3.19) 
 

功能：開啟一已下載至印表機的檔案，最多可同時開啟 2個檔案。在開啟檔案之前必須先確定

檔案已被下載至印表機的記憶體內。 

 

語法： 

OPEN “Filename”, File handle 
 

參     數   說          明 

Filename    已載至印表機的檔案名稱 

File handle    0或 1  

 

範例： 

若有一檔案名叫”DATA”的被載至印表機，且檔案內容如下： 

 
DOWNLOAD "DATA1",56,COMPUTER 

2000 
12 
MOUSE 
500 
13 
KEYBOARD 
300 
100 

   
DOWNLOAD "DATA2",56,Computer 
3000 
32 
Mouse 
900 
93 
Keyboard 
700 
700 

 

        儲存上述內容於檔名叫”DATA”的檔案之中，下述步驟可將”DATA”內的資料下載至印表機

內。 

 
C:\>COPY DATA/B LPT1 
 

若有一個名叫”DEMO.BAS”的檔案被下載，且其內容如下： 

 
DOWNLOAD "DEMO.BAS" 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
I=1 
Y=100 
GOSUB OpenData 

:Start 
CLS 
TEXT 10,10,"3",0,1,1,"*****OPEN COMMAND TEST*****" 
ITEM$="" 
READ 0,ITEM$,P,Q 
TEXT 10,Y,"2",0,1,1,ITEM$+"$"+STR$(P)+"[EOF(0)="+STR$(EOF(0))+"]" 
BARCODE 10,Y+25,"39",40,1,0,2,4,"PRICE*"+STR$(Q)+"="+STR$(P*Q) 
Y=Y+100 
PRINT 1 
Y=100 
IF EOF(0)=1 THEN GOSUB OpenData 
IF EOF(0)=0 THEN GOTO Start 
END 
:OpenData 
IF I=1 THEN OPEN "DATA1",0 
IF I=2 THEN OPEN "DATA2",0 
SEEK 0,0 
IF I>2 THEN END 
I=I+1 
RETURN 
EOP 
DEMO 
 

儲存上述資料內容並命其檔名為”DEMO”，下述步驟可將”DEMO”檔存於印表機內並執行

該檔案。 

 

<於 DOS模式下>： 

C:\>COPY DEMO/B LPT1  
 
執行印表機內”DEMO.BAS”檔案： 

C:\>COPY CON LPT1 
DEMO 
<Ctrl><Z> 
 

上述範例可將”DATA1”及”DATA2”以相同的檔案編號 (file handle) 0開啟，並讀出檔案內

的內容。 

 
其他參考項目： 

DOWNLOAD, EOP, READ, EOF, LOF, SEEK, FREAD$() 



 

TAIWAN SEMICONDUCTOR CO., LTD.          75          Copyright 2001 All Rights Reserved. 
 

 

 READ 
 

功能：讀出已開啟檔案的資料內容逐行讀出，並存於變數之中。 

 

語法： 

READ file handle, variables 
 

參     數   說          明 

file handle   0或 1 

variables    字串變數，整數變數或浮點變數 

 

範例： 

DOWNLOAD "DATA1",20,COMPUTER 
2000 
12 
 
DOWNLOAD "DATA2",16,Mouse 
900 
93 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
I=0 
Y=100 
OPEN "DATA1",0 
OPEN "DATA2",1 
SEEK 0,0 
SEEK 1,0 
:Start 
CLS 
TEXT 10,10,"3",0,1,1,"*****READ COMMAND TEST*****" 
TEXT 10,50,"3",0,1,1,"OPEN-READ DATA"+STR$(I+1) 
ITEM$="" 
READ I,ITEM$,P,Q 
TEXT 10,Y,"2",0,1,1,ITEM$+"$"+STR$(P) 
BARCODE 10,Y+25,"39",40,1,0,2,4,"PRICE*"+STR$(Q)+"="+STR$(P*Q) 
Y=Y+100 
PRINT 1 
Y=100 
IF I<=1 THEN 
  IF EOF(I)=1 THEN 
    I=I+1 
    GOTO Start 
  ELSE 
    GOTO Start 
  ENDIF 
ELSE 

  END 
ENDIF 
EOP 
DEMO 

 
其他參考項目： 

DOWNLOAD, EOP, OPEN, EOF, LOF, SEEK, FREAD$() 

 SEEK 
 

功能：用來控制某已開啟檔案內的指標位置。 

 

語法： 

SEEK file handle, offset 
 

參     數   說          明 

file handle   0 or 1 

offset    指標位置的偏移量 is  

 

範例： 

DOWNLOAD "DATA",12,1234567890 
 
DOWNLOAD "TEST.BAS" 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 1 
REFERENCE 0,0 
CLS 
OPEN "DATA",0 
SEEK 0,4 
READ 0,Num$ 
   TEXT 100,10,"3",0,1,1,"SEEK COMMAND TEST" 
BAR 100,40,300,4 
   TEXT 100,60,"3",0,1,1,"SHIFT 4 CHARACTERS" 
   TEXT 100,110,"3",0,1,1,Num$ 
BAR 100,140,300,4 
SEEK 0,0 
READ 0,Num$ 
   TEXT 100,160,"3",0,1,1,"SHIFT 0 CHARACTERS" 
   TEXT 100,210,"3",0,1,1,Num$ 
PRINT 1 
EOP 
TEST 
 

其他參考項目： 

DOWNLOAD, EOP, OPEN, READ, EOF, LOF, FREAD$() 

 LOF( ) 
 

功能：回傳指定檔案的大小。 

 

語法： 

LOF (“FILENAME”) 
 

參     數   說          明 

FILENAME   已下載至印表機的檔案 

 

範例： 

DOWNLOAD "DATA1",10,1234567890 
 

DOWNLOAD "DATA2",15,ABCDEFGHIJKLMNO 
 

DOWNLOAD "LofTest.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
  OPEN "DATA1",0 
  OPEN "DATA2",1 
  TEXT 10,20,"4",0,1,1,"LOF() FUNCTION TEST" 
  J=LOF("DATA1") 
  K=LOF("DATA2") 

    
TEXT 10,140,"3",0,1,1,"DATA1 IS: "+STR$(J)+" Bytes" 
TEXT 10,200,"3",0,1,1,"DATA2 IS: "+STR$(K)+" Bytes" 

  PRINT 1 
EOP 

 LofTest 
 
其他參考項目： 

DOWNLOAD, EOP, OPEN, READ, EOF, SEEK, FREAD$() 

 FREAD$( ) 
 

功能：讀出一檔案內指定長度的資料。 

 

語法： 

FREAD$ (file handle, byte) 
 

參     數   說          明 

file handle   0或 1 

byte     指定資料長度，以 byte計 

 

範例： 

DOWNLOAD "DATA1",10,1234567890 
 
DOWNLOAD "DATA2",15,ABCDEFGHIJKLMNO 
 
DOWNLOAD "OPEN2.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
  Y$="" 
  Z$="" 
  OPEN "DATA1",0 
  OPEN "DATA2",1 
  SEEK 0,0 
  SEEK 1,0 
  TEXT 10,260,"3",0,1,1,"FREAD$(0,6) IS: "+Y$ 
  TEXT 10,320,"3",0,1,1,"FREAD$(1,6) IS: "+Z$ 
  PRINT 1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, OPEN, READ, EOF, LOF(), SEEK 

 FOR…NEXT LOOP 
 

功能：重覆執行單一行或多行指令，必須定義重覆執行的次數，且最大可執行 39層蜂巢迴圈。

於 FOR…NEXT 迴圈執行中途跳開迴圈是不被允許的，因為那不是一個好的程式設計技

巧。 

 

語法： 

For variable = start TO end STEP increment 
statement; start < end 

NEXT 
 

參     數   說          明 

variable    變數名稱，最多 8個字元 

start     定義迴圈起始值，可為整數或浮點數 

end     定義迴圈結束值，可為整數或浮點數 

increment      定義迴圈間隔值，可為整數或浮點數 

 

範例： 

DOWNLOAD "LOOP.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 1 
CLS 
FOR I=1 TO 10 STEP 1 
TEXT 100,10+30*(I-1),"3",0,1,1,STR$(I) 
NEXT 
FOR I=1 TO 1000 STEP 100 
TEXT 200,10+((I-1)/10)*3,"3",0,1,1,STR$(I) 
NEXT 
FOR I=110 TO 10 STEP -10 
TEXT 300,10+(ABS(I-110))*3,"3",0,1,1,STR$(I) 
NEXT 
FOR I=1 TO 5 STEP 0.5 
IF I-INT(I)=0 THEN Y=10+60*(I-1) ELSE Y=Y+30 
TEXT 400,Y,"3",0,1,1,STR$(I) 
NEXT 
PRINT 1 
EOP 
LOOP 
 

其他參考項目： 

DOWNLOAD, EOP 

 IF…THEN…ELSE…ENDIF  LOOP 
 

功能：使用 IF…THEN 判斷並執行區段內單一行或多行指令，可使用單一行的語法或多行區塊

的語法。 

 

語法： 

     單行語法： 

IF condition THEN statement 
 

  注意事項：單一行語法不需使用ENDIF做結尾 

 

或區塊語法一： 

 
IF condition THEN 

Statements 
ENDIF 
 

或區塊語法二： 

 
IF condition THEN 

Statements 
ELSE 

Statements 
ENDIF 
 

或區塊語法三： 

 
IF condition 1 THEN 

Statement block 1 
ELSEIF condition 2 THEN 
  Statement block 2 

 ．．． 

ELSEIF condition n THEN 
  Statement block n 
ENDIF 

 

單行語法 IF…THEN…ELSE中指令的部份必需小於 255個字元 

 

參     數   說          明 

condition    判斷式，其中可用的關係式為：<, >, =, <=, >= 

statement      陳述句，每行僅可有一句陳述句 

 

範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.12,0 
SPEED 4 
DENSITY 8 
DIRECTION 1 
REFERENCE 0,0 
OFFSET 0.00 
SET CUTTER OFF 
SET PEEL OFF 

CLS 
 
A=0 
B=0 
C=0 
D=0 
E=0 
F=0 
G=0 
H=0 
J=0 
K=0 
L=0 
 
FOR I=1 TO 100 
  IF I-INT(I/1)*1=0 THEN A=A+I 
  IF I-INT(I/2)*2=1 THEN B=B+I ELSE C=C+I 
  IF I-INT(I/3)*3=0 THEN 
    D=D+I 
  ENDIF 
  IF I-INT(I/5)*5=0 THEN 
    E=E+I 
  ELSE 
    F=F+I 
  ENDIF 
  IF I-INT(I/7)*7=0 THEN 
    G=G+I 
  ELSEIF I-INT(I/17)*17=0 THEN 
    H=H+I 
  ELSEIF I-INT(I/27)*27=0 THEN 
    J=J+I 
  ELSEIF I-INT(I/37)*37=0 THEN 
    K=K+I 
  ELSE 
    L=L+I 
  ENDIF 
NEXT 
TEXT 100,110,"3",0,1,1,"(1) 1+2+3+...+100="+STR$(A) 
TEXT 100,160,"3",0,1,1,"(2) 1+3+5+...+99="+STR$(B) 
TEXT 100,210,"3",0,1,1,"(3) 2+4+6+...+100="+STR$(C) 
TEXT 100,260,"3",0,1,1,"(4) 3+6+9+...+99="+STR$(D) 
TEXT 100,310,"3",0,1,1,"(5) 5+10+15+...+100="+STR$(E) 
TEXT 100,360,"3",0,1,1,"    (1)-(5)="+STR$(F) 
TEXT 100,410,"3",0,1,1,"(6) 7+14+21+...+98="+STR$(G) 
TEXT 100,460,"3",0,1,1,"(7) 17+34+51+...+85="+STR$(H) 
TEXT 100,510,"3",0,1,1,"(8) 27+54+...+81="+STR$(J) 
TEXT 100,560,"3",0,1,1,"(9) 37+74="+STR$(K) 
TEXT 100,610,"3",0,1,1,"    (1)-(6)-(7)-(8)-(9)="+STR$(L) 
PRINT 1,1 
EOP 
 
DOWNLOAD “IFTHEN.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 3 

DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
A=50 
B=5 
C$="" 
D$="" 
 
:L1 
IF A>100 THEN GOTO L1 ELSE A=A+10 
C$=STR$(A)+" IS SMALLER THAN 100" 
TEXT 100,10,"4",0,1,1,C$ 
PRINT 1 
END 
 
:L2 
A=A+B 
D$=STR$(A)+" IS LARGER THAN 100" 
TEXT 100,100,"4",0,1,1,D$ 
PRINT 1 
GOTO L1 
EOP 

注意事項： 

單行語法：當IF後的判斷式為真時，執行THEN後方的陳述句；為假時，執行ELSE後方的
陳述句。若無ELSE，則程式將執行IF…THEN之後的陳述句。 

區塊語法：必須以ENDIF結尾 

限制：一個程式中IF…THEN…ELSE最多僅可使用40組 

 
其他參考項目： 

DOWNLOAD, EOP 

 GOSUB…RETURN 
 

功能：跳至某一標籤(副程式)執行，執行完成後遇到”RETURN”時再跳回原位置繼續執行下一

行程式。 

 

注意事項：一個程式中 GOSUB…RETURN最多僅可使用 40組 

 

語法： 

GOSUB LABEL 
statement 

END 
:LABEL 

statement 
RETURN 
 

參     數   說          明 

LABEL    副程式的開端，標籤名稱最長度 8個字元 

 

範例： 

DOWNLOAD "GOSUB1.BAS" 

SIZE 3,3 

GAP 0,0 

DENSITY 8 

SPEED 4 

DIRECTION 0 

CLS 

TEXT 10,10,"3",0,1,1,"GOSUB & RETURN COMMAND TEST" 

GOSUB DR_BOX 

PRINT 1 

END 

:DR_BOX 

  FOR I=21 TO 81 STEP 10 

      BOX 80+I,80+I,80+300-I,80+300-I,5 

  NEXT 

RETURN 

EOP 

GOSUB1 

 
其他參考項目： 

DOWNLOAD, EOP, END, GOTO 

 GOTO 
 

功能：跳至某一標籤位置往下執行，標籤名稱最長 8個字元 

 

 注意事項：一程式中標籤數(LABEL)最多僅可有 40組 

 

語法： 

GOTO LABEL 
 
:LABEL 
statement 
 

範例： 

DOWNLOAD "GOTO1.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 1 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
A=0 
TOTAL=0 
:START 
  IF A<100 THEN  
    GOTO SUM  
  ELSE  
    GOTO PRTOUT 
  ENDIF 
:SUM 
  A=A+1 
  TOTAL=TOTAL+A 
  GOTO START 
:PRTOUT 
  B$="THE SUMMATION OF 1..100 IS "+STR$(TOTAL) 
  TEXT 10,100,"3",0,1,1,B$ 
  PRINT 1 
END 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, END, GOSUB…RETURN 

 INP$( ) 
 

    功能：透過串列埠接收字元，每次僅能接收一個字元 

 

語法： 

INP$(n) 
 

參     數   說          明 

n     1: 經由印表機的 com1端口接收 

 

範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
T$="" 
FOR I=1 TO 5 
  T$=T$+INP$(1) 
NEXT 
TEXT 100,100,"4",0,1,1,"INP$(1)="+T$ 
PRINT 1 
EOP 
DEMO 
12345 

 
其他參考項目： 

DOWNLOAD, EOP, END, INPUT, GOSUB…RETURN, GOTO 

 INPUT 
 

    功能：透過串列埠接收資料，通常用於可攜帶式鍵盤 KP-200 

 

語法： 

INPUT [“Prompt string”, number of digits], variables 
The comma also can be replaced by semicolon. Such as: 
INPUT [“Prompt string”; number of digits]; variables 
 

參     數   說          明 

Prompt string  顯示於 KP-200 LCD 上的提示字元，最大長度為 20個字元 

Number of digits  輸入資料長度限制，最大長度為 255個字元 

Variables     接收資料用的變數 

 

範例： 

DOWNLOAD "INPUT1.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
 
:START 
  CLS 
  A$="" 
  TEXT 20,50,"3",0,1,1,"INPUT and KP-200 Test" 
  INPUT "CODE 39 :",C39$ 
  INPUT "EAN 13:",12,E13$ 
 
  BARCODE 20,100,"39",48,1,0,2,5,C39$ 
  BARCODE 20,200,"EAN13",48,1,0,4,4,E13$ 
 
  PRINT 1 
GOTO START 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, END, GOTO 

 REM 
 

    功能：註解用，於字首加註”REM”則該行將被忽略不執行。 

 

語法： 

REM string 
 

範例： 

REM ******************************** 
REM This is a demonstration program* 
REM ********************************  
DOWNLOAD "REMARK.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 1 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 50,50,"3",0,1,1,"REMARK DEMO PROGRAM" 
REM TEXT 50,100,"3",0,1,1,"REMARK DEMO PROGRAM" 
PRINT 1,1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, END 

 OUT 
 

    範例：經印表機的串列埠透過 RS-232送出資料，使用 KP-200時必須搭配”~#…~&”使用 

 

語法： 

OUT “prompt”, variable 
 

參     數         說          明 

prompt    顯示於 KP-200 LCD 上的提示字元 

Variable    透過 RS-232送出的資料 

 

範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
PRICE$="123456" 
OUT "PRICE:",PRICE$ 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, END, ~#…~& 

 GETKEY( ) 
 

功能：取得 PAUSE鍵及 FEED鍵的狀態，此函數會持續等待直到按鍵被按為止。當 PAUSE鍵

被按時回傳值為 0，FEED 鍵被按時回傳值為 1。 

 

機   種 PAUSE FEED 

TTP246M/344M 0 1 

TTP-245/343/TDP-245 X 1 

 

 語法： 

GETKEY() 
 

範例： 

DOWNLOAD “DEMO4.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
:START 
A=GETKEY() 
IF A=0 THEN GOTO PAUSEB 
IF A=1 THEN GOTO FEEDB 
:PAUSEB 
CLS 
TEXT 50,10,"4",0,1,1,"PAUSE key is pressed !" 
PRINT 1 
GOTO START 
:FEEDB 
CLS 
TEXT 50,10,"4",0,1,1,"FEED key is pressed !" 
PRINT 1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, END, GOTO 

 INT( ) 
 

功能：回傳浮點數的整數部份，小數部份則刪去。 

 

語法： 

INT (n) 
 

參     數   說          明 

n     n可以是正整數、負整數、浮點數或數學運算式  

 

範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 4,2 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
REM **** To round up or down**** 
INPUT "Number:",Num 
N=INT(Num+0.5) 
IF N>Num THEN 
  TEXT 50,100,"3",0,1,1,"To round up= "+STR$(N) 
ELSE 
  TEXT 50,100,"3",0,1,1,"To round down= "+STR$(N) 
ENDIF 
PRINT 1 
EOP 

 
其他參考項目： 

DOWNLOAD, EOP, END, ABS(), ASC(), STR$() 



 

TAIWAN SEMICONDUCTOR CO., LTD.          92          Copyright 2001 All Rights Reserved. 
 

 

 LEFT$( ) 
 

功能：此函數可回傳一字串中由左端向右指定長度的字串。 

  

 語法： 

LEFT$ (X$, n) 
 

參     數   說          明 

X$     待處理的字串 

n     自左端向右算起要回傳的字串長度 

 

範例： 

DOWNLOAD "STR1.BAS" 
SIZE 3.00,3.00 
GAP 0.08,0.00 
SPEED 4.0 
DENSITY 8 
SET CUTTER OFF 
DIRECTION 0 
REFERENCE 0,0 
CLS 
A$="BARCODE PRINTER DEMO PRINTING" 
C$=LEFT$(A$,10) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,100,"3",0,1,1,"10 LEFT 10 CHARS: "+C$ 
PRINT 1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, END, RIGHT$(), MID$(), LEN(), STR$() 

 LEN( ) 
 

功能：回傳一字串的長度 

語法： 

LEN (string) 
 

參     數   說          明 

string    待回傳長度的字串 

 

範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0.08,0.00 
SPEED 4.0 
DENSITY 8 
SET CUTTER OFF 
DIRECTION 0 
REFERENCE 0,0 
CLS 
A$="TAIWAN SEMICONDUCTOR CO., LTD" 
B=LEN(A$) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,50,"3",0,1,1,"STRING LENGTH="+STR$(B) 
PRINT 1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, END, LEFT$(), LEN(), RIGHT$(), MID$(), STR$(), VAL() 

 MID$( ) 

 語法： 

MID$(string, m, n) 
 

參     數   說          明 

string    待處理的字串 

m     自字串的第m 個字元開始選取，其中 1≦m≦字串長度 

n     回傳的字串長度 

 

 範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0.08,0.00 
SPEED 4.0 
DENSITY 8 
SET CUTTER OFF 
DIRECTION 0 
REFERENCE 0,0 
 
CLS 
A$="TAIWAN SEMICONDUCTOR CO., LTD" 
E$=MID$(A$,11,10) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,200,"3",0,1,1,"10 MIDDLE CHARS: "+E$ 
PRINT 1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, END, LEFT$(), LEN(), RIGHT$(), STR$(), VAL() 

 RIGHT$( ) 
 

功能：此函數可回傳一字串中由右端向左指定長度的字串。 

  

 語法： 

RIGHT$ (X$, n) 
 

參     數   說          明 

X$     待處理的字串 

n     自右端向左算起要回傳的字串長度 

 

 範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0.08,0.00 
SPEED 4.0 
DENSITY 8 
SET CUTTER OFF 
DIRECTION 0 
REFERENCE 0,0 
 
CLS 
A$="TAIWAN SEMICONDUCTOR CO., LTD" 
D$=RIGHT$(A$,10) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,150,"3",0,1,1,"10 RIGHT CHARS: "+D$ 
PRINT 1 
EOP 
 

其他參考項目： 

DOWNLOAD, EOP, END, LEFT$(), LEN(), MID$(), STR$(), VAL() 

 STR$( ) 
 

功能：將某一數值轉換為字串格式 

 

 語法： 

STR$ (n) 
 

參     數   說          明 

n     可為一整數、浮點數或數學運算式 The 

  

 範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0,0.00 
SPEED 4.0 
DENSITY 8 
SET CUTTER OFF 
DIRECTION 0 
REFERENCE 0,0 
CLS 
A$="TAIWAN SEMICONDUCTOR CO., LTD" 
F=100 
G=500 
H$=STR$(F+G) 
TEXT 10,10,"3",0,1,1,A$ 
TEXT 10,60,"3",0,1,1,"F="+STR$(F) 
TEXT 10,110,"3",0,1,1,"G="+STR$(G) 
TEXT 10,160,"3",0,1,1,"F+G="+H$ 
PRINT 1 
EOP 
DEMO 

 
 

其他參考項目： 

DOWNLOAD, EOP, END, LEFT$(), LEN(), RIGHT$(), MID$(), VAL() 

 VAL( ) 
 

功能：數值字串轉換為整數型態或浮點數型態 

 

 語法： 

VAL (“numeric character”) 
 

參     數   說          明 

numeric character  數字 0~9及小數點“.” 

 

範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3.00,3.00 
GAP 0.00,0.00 
SPEED 4.0 
DENSITY 8 
SET CUTTER OFF 
DIRECTION 0 
REFERENCE 0,0 
CLS 
 A$="TAIWAN SEMICONDUCTOR CO., LTD" 
 F$="100" 
 G$="500" 
 H=VAL(F$)+VAL(G$) 
 I$=STR$(H) 
 TEXT 10,10,"3",0,1,1,A$ 
 TEXT 10,60,"3",0,1,1,"F="+F$ 
 TEXT 10,110,"3",0,1,1,"G="+G$ 
 TEXT 10,160,"3",0,1,1,"F+G="+I$ 
 PRINT 1 
EOP 
DEMO 

 
 

其他參考項目： 

DOWNLOAD, EOP, END, LEFT$(), LEN(), RIGHT$(), MID$(), STR$() 

 BEEP 
 

功能：印表機會送出 0x07至 KP-200使其發出嗶聲 

 

語法： 

BEEP 
 

範例： 

 
DOWNLOAD “DEMO.BAS” 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
BEEP 
INPUT “Text1 =”,TEXT1$ 
TEXT 100,100,”3”,0,1,1,TEXT1$ 
PRINT 1 
EOP 

Device Reconfiguration Commands 
 SET COUNTER  
 

功能：COUNTER 可以是一般的計數器或是一變數。可設定計數器的初始值及計數使用的增量，

且計數器有三種型式：數值型式(即 0~9~0)，小寫英文字母型式(a~z~a)及大寫英文字母

型式(A~Z~A) 

 

 語法： 

SET COUNTER @n step  
@n = “Expression” 
 

參     數   說          明 

@n  n: 計數器的初始值。總計有 51(@0~@50)個計數器可同時使

用  

step     計數器的增量值可為正數亦可為負數。 

-999999999<= step <=999999999 

如果計數器只是被當做固定的變數使用時，請將增量值設為 0 

Expression   初始字串，字串最長 101bytes 

 

 範例： 

SIZE 3,3 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET COUNTER @1 1 
@1="00001" 
SET COUNTER @2 5 
@2="AB000001" 
CLS 
TEXT 50,50,"3",0,1,1,@1 
BARCODE 50,100,"39",48,1,0,2,4,@2 
PRINT 2,1 
 

其他參考項目： 

PRINT, TEXT, BARCODE 

 SET CUTTER 
 
功能：此設定用來開啟/關閉裁刀裁切的功能，並可定義每印幾張後做一次裁切動作。當印表機

電源關閉時，此項設定會被記錄於 EEPROM。 

 

 語法： 

SET CUTTER OFF/BATCH/pieces 
 

參     數   說          明 

OFF     關閉裁刀 

BATCH 設定在列印結束時才執行裁切動作 

Pieces    設定每印幾張後做一次裁切動做 0<= pieces <=65535 

 

 範例： 

REM ***SET CUTTER FUNCTION OFF EXAMPLE PROGRAM*** 
SIZE 3,3 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 50,50,"3",0,1,1,"SET CUTTER OFF" 
PRINT 3 
REM ***This program cuts once at the batch*** 
SET CUTTER BATCH 
CLS 
TEXT 50,50,"3",0,1,1,"SET CUTTER BATCH" 
PRINT 3,2 
REM ***This program cuts every label*** 
SET CUTTER 1 
CLS 
TEXT 50,50,"3",0,1,1,"SET CUTTER 1" 
PRINT 3,2 
REM ***This program cuts 2 label*** 
SET CUTTER 2 
CLS 
TEXT 50,50,"3",0,1,1,"SET CUTTER 2" 
PRINT 3,2 
 

其他參考項目： 

OFFSET, PRINT, SET PARTIAL_CUTTER 

 SET PARTIAL_CUTTER (New item by V3.18)  
 
功能：此設定用來開啟/關閉裁刀半切的功能，並可定義每印幾張後做一次裁切動作。當印表機

電源關閉時，此項設定會被記錄於 EEPROM。所謂的半切即指當裁切完後標籤紙並不

會回拉至列印線位置，於下一張列印時會直接就該位置列印，故所得列印後的標籤其前

端會有一段空白。 

 

 語法： 

SET PARTIAL_CUTTER OFF/BATCH/pieces 
 

參     數   說          明 

OFF     關閉裁刀半裁功能 

BATCH 設定在列印結束時才執行半切動作 

Pieces    設定每印幾張後做一次裁切動做 s0<= pieces <=65535 

 

 範例： 

REM **SET PARTIAL_CUTTER FUNCTION OFF EXAMPLE PROGRAM** 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET PARTIAL_CUTTER OFF 
CLS 
TEXT 50,50,"3",0,1,1,"SET PARTIAL_CUTTER OFF" 
PRINT 3 
REM ***This program cuts once at the batch*** 
SET PARTIAL_CUTTER BATCH 
CLS 
TEXT 50,50,"3",0,1,1,"SET PARTIAL_CUTTER BATCH" 
PRINT 3,2 
REM ***This program cuts every label*** 
SET PARTIAL_CUTTER 1 
CLS 
TEXT 50,50,"3",0,1,1,"SET PARTIAL_CUTTER 1" 
PRINT 3,2 
REM ***This program cuts 2 label*** 
SET PARTIAL_CUTTER 2 
CLS 
TEXT 50,50,"3",0,1,1,"SET PARTIAL_CUTTER 2" 
PRINT 3,2 
 

其他參考項目： 

OFFSET, PRINT, SET CUTTER 

 SET BACK (New item by V3.18)  
 
功能：此項設定可使印表機列印及裁切結束時不執行回拉的動作，通常置於 SET CUTTER 功

能之後 

 

 語法： 

SET BACK OFF/ON 
 

參     數   說         明 

OFF     關閉裁切後標籤紙被回拉至列印線位置的功能 

ON 開啟裁切後標籤紙被回拉至列印線位置的功能 

 

 範例： 

REM **SET BACK FUNCTION OFF EXAMPLE PROGRAM** 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 1 
REFERENCE 0,0 
SET CUTTER 1 
SET BACK OFF 
CLS 
TEXT 50,50,"3",0,1,1,"SET BACK OFF" 
PRINT 3 
 
CLS 
SET CUTTER 1 
SET BACK ON 
TEXT 50,50,"3",0,1,1,"SET BACK ON" 
PRINT 3 
 

其他參考項目： 

OFFSET, PRINT, SET CUTTER 

 SET KEY1, SET KEY2, SET KEY3 
 

功能：設定開啟/關閉KEY1/KEY2/KEY3的功能。KEY1預設為”MENU”鍵，KEY2預設為”PAUSE”

鍵及 KEY3 預設為 ”FEED”鍵。在設定 KEY1/KEY2/KEY3 的功能之前，必須先將

KEY1/KEY2/KEY3 的預設功能先關閉，當印表機關閉電源時所重新設定的

KEY1/KEY2/KEY3的功能將被記錄於印表機的 EEPROM中。 

  

印表機種類 KEY1 KEY2 KEY3 

TTP-246M/344M V V V 

TTP-245/343/TDP-245 V X X 

 

 語法： 

SET KEY1 ON/OFF 
SET KEY2 ON/OFF 
SET KEY3 ON/OFF 
 

參     數   說          明 

ON     開啟 KEY1定義為MENU鍵的功能 

      開啟 KEY2定義為 PAUSE鍵的功能 

      開啟 KEY3定義為 FEED鍵的功能 

OFF     關閉 KEY1定義為MENU鍵的功能 

      關閉 KEY2定義為 PAUSE鍵的功能 

      關閉 KEY3定義為 FEED鍵的功能 

 

注意事項：設定值將會被記錄於印表機的 EEPROM中 

 

 範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3,1 
GAP 0,0 
DENSITY 8 
SPEED 3 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET KEY1 OFF 
SET KEY2 OFF 
SET KEY3 OFF 
KEY1=0 
KEY2=0 
KEY3=0 
 
CLS 
:START 
 
IF KEY1=1 THEN 
  CLS 
  TEXT 100,10,"3",0,1,1,"KEY1 (MENU key) is pressed!!" 
  PRINT 1,1 
ELSEIF KEY2=1 THEN  
  CLS 
  TEXT 100,10,"3",0,1,1,"KEY2 (PAUSE key) is pressed!!" 
  PRINT 1,1 

ELSEIF KEY3=1 THEN 
  CLS 
  TEXT 100,10,"3",0,1,1,"KEY3 (FEED key) is pressed!!" 
  TEXT 100,60,"3",0,1,1,"End of test" 
  PRINT 1,1 
  SET KEY1 ON 
  SET KEY2 ON 
  SET KEY3 ON 
  END 
ENDIF 
GOTO START 
EOP 
DEMO 
 

其他參考項目： 

OFFEST, PRINT 

 SET LED1, SET LED2, SET LED3 
 

功能：此項設定通常用來控制印表機 LED燈號的功能，LED1、LED2及 LED3的燈號預設值如

下： 

 

印表機型號 LED1 LED2 LED3 LED2 & LED3 

TTP246M/344M V V V  

TTP245/343/TDP245 Green Green Red Orange 

注意事項：TTP-245/343/TDP-245的 LED1燈號與 LED2燈號相同。 

 

LED no.   預 設 功 能 

LDE1   電源燈號 

LED2   待機燈號 

LED3   錯誤燈號 

 

 語法： 

SET LED1 ON/OFF 
SET LED2 ON/OFF 
SET LED3 ON/OFF 
 

 範例： 

DOWNLOAD "DEMO4.BAS" 
SET LED1 OFF 
SET LED2 OFF 
SET LED3 OFF 
FOR I=1 TO 100 
LED1=0 
LED2=0 
LED3=0 
  IF I-INT(I/2)*2=0 THEN 
    LED1=1 
  ELSEIF I-INT(I/3)*3=0 THEN 
    LED2=1 
  ELSE 
    LED3=1 
  ENDIF 
NEXT 
LED1=1 
LED2=1 
LED3=0 
SET LED1 ON 
SET LED2 ON 
SET LED3 ON 
EOP 
DEMO4 

 SET PEEL 
 

功能：設定啟動/關閉自動剝紙器功能。預設值為關閉狀態，當此功能被開啟時，印表機會在每

印完一張時即暫停，直到標籤紙被取走後才會印出下一張標籤。此項設定會被記錄於印

表機的 EEPROM中。 

 

 語法： 

SET PEEL ON/OFF 
 

參     數   說          明 

ON     開啟自動剝紙器的功能 

OFF     關閉自動剝紙器的功能 

 

 範例： 

REM ***SELF-PEELING FUNCTION ON*** 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL ON 
CLS 
TEXT 50,100,"3",0,1,1,"SELF-PEELING FUNCTION TEST" 
PRINT 5 
 

其他參考項目： 

OFFEST, PRINT 

 SET TEAR 
 

功能：設定開啟/關閉送紙至撕紙線的功能，此項設定會被記錄於印表機內的 EEPROM中 

 

 語法： 

SET TEAR ON/OFF 
 

參     數   說          明 

ON 標籤列印結束時將送紙至撕紙位置 

OFF 標籤列印結束時會將標籤起印點停留至列印線位置 

 

 範例： 

REM ***TEAR FUNCTION ON*** 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
SET TEAR ON 
CLS 
TEXT 50,100,"3",0,1,1,"TEAR FUNCTION TEST" 
PRINT 1 
 

其他參考項目： 

SET PEEL, SET CUTTER 

 SET GAP  
 

功能：用來設定 GAP感應器的發射強度。GAP感應器校正時，通常都是以按住 PAUSE鍵後開

啟印表機電源來完成校正 GAP感應器強度的校正，但有時標籤底紙的厚度或是標籤本身

含有浮水印時均有可能造成 GAP感應器的誤判。在這種情形下，使用者必須使用手動方

式對 GAP感應器實施校正動作，透過反覆試驗後即可獲得適當的感應器設定值。此項設

定值將被記錄於印表機內的 EEPROM中。 

 

 語法： 

SET GAP n/AUTO/OFF/0,/REVERSE/OBVERSE 
 

參     數   說          明 

n  GAP感應器發射強度值，下方表格中說明了各機種 GAP感應

器強度的範圍，0為最小強度。 

AUTO               於每次開機時或是開闔印表頭時，印表機均會自動向外送出2到3張

標籤紙對GAP感應器重新校正一次。若遇到連續紙時，標籤紙會被

向外送約10~20吋的長度以確定紙張是否為連續紙。 

OFF 取消SET GAP AUTO的功能。 

0,                   於待機狀態下，自動執行GAP感應器校正。 

REVERSE           此項功能是針對部份以標籤紙於紙張正面以黑標做為標籤之間的

區隔使時，使用此種紙張時無法以黑標感應器實施校正，故藉由將

GAP感應器反相的動作來完成對此類型紙張的校正動作。(所謂的

反相是指：原GAP感應器的定義為穿透的部份為間隙，未穿透的部

份為紙張；反相後則定義穿透的部份為紙張，未穿透的部份為黑標

間隙) 

OBVERSE           取消SET GAP REVERSE的功能。 

 

印表機型號 
感應器強

度範圍 

SET GAP REVERSE 
SET GAP OBVERSE 

SET GAP AUTO 

TTP-2410M 0~255 X 
TTP-248M 0~255 X 
TTP-246M/344M series (non usb) 0~63 X 
TTP-246M/344M series (usb) 0~31 X 
TTP-245/343 series 0~63 V (自 V3.19) 

TDP-245 0~63 V (自 V3.19) 

 

注意事項：於關閉”印字頭開啟的感應器”(即 SET HEAD OFF)的模式下，則 SET GAP 

AUTO的動作於開闔印字頭時並不會自動執行校正動作，但於電源開啟時校正
動作仍會被執行。 

 

 範例： 

下方範例是在 DOS 模式底下透過並列埠(Parallel port)對印表機設定標籤尺寸及標籤間隙

高度與 GAP感應器的強度。 

 
C:\>COPY CON LPT1<ENTER> 

SIZE 4,2.5<ENTER> 
GAP 0.12,0<ENTER> 
SET GAP 1<ENTER> 
<CTRL><Z><ENTER> 

C:\> 
 

注意事項：上方範例中<ENTER>表示按下鍵盤上的”ENTER”鍵，<CTRL>表示按下鍵
盤上的”Ctrl”鍵，<Z>表示按下鍵盤上的”Z”鍵。 

 

 可藉由按 FEED 鍵來測試 GAP感應器校正是否正確，每按一次 FEED鍵標籤紙停留的位

置是否均一致且錯誤燈號不閃爍。若否，則稍微大幅度的調整感應器強度再試一次，測試

時請由 0開始逐漸增加強度。 

 
其他參考項目： 

SIZE, GAP, BLINE 

 SET HEAD  
 

功能：設定開啟/關閉印字頭開啟感應器。若感應器為關閉狀態，則當印字頭被開啟時將不會有

任何錯誤訊息出現，且該設定會被存於印表機的 EEPROM中。 

 

 語法： 

SET HEAD ON /OFF 
 

參     數   說          明 

ON     開啟印字頭感應器 

OFF     關閉印字頭感應器 

 

 範例： 

SET HEAD ON 
SET HEAD OFF 

 SET RIBBON  
 

功能：設定開啟/關閉碳帶感應器，即切換熱轉式/熱感印式列印。通常印表機於開啟電源時，

碳帶感應器即會自動偵測印表機是否已裝上碳帶，並藉此決定使用熱感式或熱轉式列

印。此項設定並不會存於 EEPROM中。 

 

 語法： 

SET RIBBON ON /OFF 
 

參     數   說          明 

ON     熱轉式列印 

OFF     熱感式列印 

 

 範例： 

REM ***Direct printing**** 
SIZE 4,4 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
SET RIBBON OFF 
CLS 
BARCODE 100,100,"39",48,1,0,2,5,"CODE 39” 
PRINT 1 

 SET COM1 
 

功能：設定印表機串列埠的通訊參數。 

 

 語法： 

SET COM1 baoud, parity, data, stop 
 

參     數   說          明 

boaud        鮑率，可用的鮑率速度如下： 

24: 2400 bps 
48: 4800 bps 
96: 9600 bps 
19: 19200 bps 

 

parity    同位檢查 

      N: 無 

      E: 偶數 

      O: 奇數 

 

data     資料位元 

      8: 8 位元資料 

      7: 7 位元資料 

 

stop     停止位元 

      1 
      2 
 

 範例： 

通常於 DOS下透過並列埠(Parallel port)設定印表機的串列埠通訊參數。 

 
C:\>COPY CON LPT1<ENTER> 
SET COM1 19,N,8,1<ENTER> 
<CTRL><Z><ENTER> 
C:\> 
 

注意事項：上方範例中<ENTER>表示按下鍵盤上的”ENTER”鍵，<CTRL>表示按下鍵
盤上的”Ctrl”鍵，<Z>表示按下鍵盤上的”Z”鍵。 

 SET PRINTKEY (by V3.03) 
 

功能：此設定會印出一張標籤，並將標籤前推至撕紙位置。當按下 FEED 鍵時會印出下一張標

籤內容或是下一批某數量的標籤。若標籤內容含有連續跳號的文字或條碼時，跳號的部

份將會依序改變，此項設定將被記錄於 EEPROM中。 

 

 語法： 

 SET PRINTEKY OFF/ON/AUTO/<num> 
 

參     數   說          明 

OFF     關閉此項功能 

oON        開啟此項功能 

oAUTO       開啟此項功能 

<num>    設定當按下 FEED鍵時會印出的張數 

 

範例： 

先執行下方程式： 

SIZE 4,2.5 
GAP 0.12,0 
SET FEEDPRINT ON 
SET COUNTER @0 1 
@0="0001" 
CLS 
TEXT 10,10,"5",0,1,1,@0 
PRINT 1 

 

再執行下列動作： 

語     法 印表機接獲指令 “PRINT m” 印出結果 

SET PRINTKEY ON or 
SET PRINTKEY AUTO 

1.) PRINT 2 Label 1~2 

2.) 按下 FEED鍵 Label 3~4 

 

語     法 印表機接獲指令 “PRINT m,n” 印出結果 

SET PRINTKEY ON or 
SET PRINTKEY AUTO 

1.) PRINT 1,2 Label 1, Label 1 

2.) 按下 FEED鍵 Label 2, Label 2 

 

語    法 印表機接獲指令 “PRINT -1,n” 印出結果 

SET PRINTKEY ON or 
SET PRINTKEY AUTO 

1.) PRINT -1,2 Label 1, Label 1 

2.) 按下 FEED鍵 Label 1, Label 1 

        

語    法 印表機接獲指令 “PRINT m” 印出結果 

SET PRINTKEY 5 
1.) PRINT 2 Label 1~2 

2.) 按下 FEED鍵 Label 3~7 

 
 

語    法 印表機接獲指令 “PRINT m,n” 印出結果 

SET PRINTKEY 5 
1.) PRINT 1,2 Label 1, Label 1 

2.) 按下 FEED鍵 Label 2~6 

語    法 印表機接獲指令 “PRINT –1,n” 印出結果 

SET PRINTKEY 5 
1.) PRINT -1,2 Label 1, Label 1 

2.) Press FEED key Label 1, Label 1 

 SET REPRINT (New item by V3.19) 
 

功能：設定當發生”缺紙”、”卡紙”、”缺碳帶”或”印字頭開啟”等錯誤時，於回覆正常狀態時是否

要重印上一筆未列印完成的資料。 

 

 語法： 

SET REPRINT OFF/ON 
 

參     數   說          明 

OFF     取消重印機制 

oON        啟動重印機制 

 

範例： 

SET REPRINT ON 

 PEEL (之前未加入manual) 
 

功能：獲得剝紙感應器的狀態，狀態值僅可唯讀。 

 

 語法： 

      PEEL 
 

回  傳  值    說          明 

0                     剝紙感應器下方無紙 

1      剝紙感應器下方有紙 

 

 範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 4,1 
GAP 0,0 
SPEED 4 
DENSITY 8 
SET PEEL OFF 
SET KEY1 OFF 
SET LED1 OFF 
SET LED3 OFF 
:START 
LED1=0 
LED3=0 
  IF KEY1=1 THEN GOTO A 
GOTO START 
:A 
LED1=1 
CLS 
TEXT 10,10,"3",0,1,1,"PEEL Function Test!!" 
PRINT 1,1 
 
:B 
LED1=0 
IF PEEL=1 THEN  
  LED3=1 
  GOTO B  
ELSE 
  CLS 
  TEXT 10,10,"3",0,1,1,"The label is removed from the PEEL sensor!!" 
  PRINT 1,1 

  GOTO START 
ENDIF 
EOP 
DEMO 

 LED1, LED2, LED3 (之前未加入 manual) 
 

功能：用來控制開/關 LED 燈號，使用者僅可給定 LED 的狀態值。1 表示開啟 LED 燈號，0 表

示關閉 LED燈號。在使用此項設定時，請先取消原 LED燈號的預設值。有關各機種的燈

號設定，請參閱下表。 

 

印表機型號 LED1 LED2 LED3 LED2 & LED3 

TTP246M/344M V V V  

TTP245/343/TDP245 綠燈 綠燈 紅綠 橘燈 

注意事項：TTP-245/343/TDP-245 的 LED1與 LED2的燈號相同 

 

 語法： 

  LEDm=n 
 

參     數    說          明 

m                    m=1, LED1 
m=2, LED2 
m=3, LED3 

n      0: LED燈亮 

       1: LED燈滅 

 

 範例： 

  DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.12,0 
SPEED 4 
DENSITY 8 
DIRECTION 1 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
SET LED1 OFF 
SET LED2 OFF 
SET LED3 OFF 

  LED1=0 
LED2=1 
LED3=0 
EOP 

 KEY1, KEY2, KEY3 (之前未加入 manual) 
 

功能：用來取得 KEY1、KEY2及 KEY3的狀態值。 

 

印表機型號 KEY1 KEY2 KEY3 

TTP-246M/344M V V V 

TTP-245/343/TDP-245 V X X 

 

 語法： 

  KEYm=n 
 

按     鍵    回     傳     值 

KEY1 (MENU)   0: 未按住 

1: 按  住 

KEY2 (PAUDE)    0: 未按住 

1: 按  住 

KEY3 (FEED)    0: 未按住 

1: 按  住 

 

 範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 3,1 
GAP 0,0 
SPEED 4 
DENSITY 8 
DIRECTION 1 
REFERENCE 0,0 
SET LED1 OFF 
SET KEY1 OFF 
LED1=0 
:START 
IF KEY1=1 THEN 
 LED1=1 
 CLS 
 TEXT 100,10,"3",0,1,1,"KEY FUNCTION TEST" 
 PRINT 1,1 
ELSE 
 LED1=0 
ENDIF 
GOTO START 
EOP 
DEMO   

Printer Global Variables 
 @LABEL 
 

功能：用來計算印表機總共列印的張數。會記錄於印表機的 EEPROM中，且關開機時不會歸

零。 

 

 語法： 

寫入格式： @LABEL=n 或 @LABEL=”n” 

讀出格式：A=LABEL 或 A$=STR$(LABEL) 

 

參     數   說          明 

n      已列印的標籤數量，其中 0<=n<=999999999 

 

 範例： 

DOWNLOAD "DEMO.BAS" 
SIZE 4,2.5 
GAP 2 mm,0 
SPEED 6 
DENSITY 12 
CLS 
TEXT 10,50,"3",0,1,1,@LABEL 
TEXT 10,100,"3",0,1,1,"@LABEL="+STR$(LABEL) 
TEXT 10,150,"3",0,1,1,"*****Statement 1*****" 
  IF LABEL>1000 THEN 
    TEXT 10,200,"3",0,1,1,"LABEL>1000" 
  ELSE 
    TEXT 10,200,"3",0,1,1,"LABEL<1000" 
  ENDIF 
TEXT 10,250,"3",0,1,1,"*****Statement 1*****" 
  A=LABEL 
  IF A>1000 THEN 
     TEXT 10,300,"3",0,1,1,"A>1000" 
  ELSE 

TEXT 10,300,"3",0,1,1,"A<1000" 
  ENDIF 
TEXT 10,350,"3",0,1,1,"*****Statement 3*****" 
  A$=STR$(LABEL) 
  IF VAL(A$)>1000 THEN 
    TEXT 10,400,"3",0,1,1,"VAL(A$)>1000" 
  ELSE 
    TEXT 10,400,"3",0,1,1,"VAL(A$)<1000" 
  ENDIF 
PRINT 1,1 
EOP 



 

TAIWAN SEMICONDUCTOR CO., LTD.          120          Copyright 2001 All Rights Reserved. 
 

 

 YEAR 
 

功能：對 RTC讀出或寫入時間”年”的值，以 4位數字讀出，2位數或 4位數寫入。 

 

 語法： 

寫入格式：YEAR=02 

讀出格式：A=YEAR 

範圍：00~50=2000~2050 ; 51~99=1951~1999 

 
 

 範例： 

DOWNLOAD "SetYear.BAS" 
REM *****Set Year Parameter to RTC***** 
YEAR=05 
EOP 
SetYear 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
 
REM *****Read YEAR parameter form RTC***** 
YEAR$=STR$(YEAR) 
Y=YEAR 
 
REM *****Print***** 

TEXT 10,10,"5",0,1,1,"YEAR1="+YEAR$ 
TEXT 10,110,"5",0,1,1,"YEAR2="+STR$(Y) 
TEXT 10,210,"5",0,1,1,"YEAR3="+STR$(YEAR) 
PRINT 1 
EOP 
DEMO 

 
其他參考項目： 

~!C, MONTH, DATE, DAY, HOUR, MINUTE, SECOND 

 MONTH 
 

功能：對 RTC讀出或寫入時間”月”的值，以 2位數字(01~12)讀出或寫入。 

 

 語法： 

寫入格式：MONTH=01 

讀出格式：A=MONTH 

範圍：01~12 

 

 範例： 

DOWNLOAD "SetMonth.BAS" 
REM *****Set Month Parameter to RTC***** 
MONTH=05 
EOP 
SetMonth 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 

CLS 
 
REM *****Read Month parameter form RTC***** 
MONTH$=STR$(MONTH) 
M=MONTH 
 
REM *****Print***** 
TEXT 10,10,"5",0,1,1,"MONTH1="+MONTH$ 
TEXT 10,110,"5",0,1,1,"MONTH2="+STR$(M) 
TEXT 10,210,"5",0,1,1,"MONTH3="+STR$(MONTH) 
PRINT 1 
EOP 
DEMO 

  
其他參考項目： 

~!C, MONTH, DATE, DAY, HOUR, MINUTE, SECOND 

 DATE 
 

功能：對 RTC讀出或寫入時間”日”的值，以 2位數字(01~31)讀出或寫入。 

 

 語法： 

寫入格式：DATE=12 

讀出格式：A=DATE 

範圍：01~31 

 

 範例： 

DOWNLOAD "SetDate.BAS" 
REM *****Set Date Parameter to RTC***** 
DATE=30 
EOP 
SetDate 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
 
REM *****Read Date parameter form RTC***** 
DATE$=STR$(DATE) 
D=DATE 
 
REM *****Print***** 
TEXT 10,10,"5",0,1,1,"DATE1="+DATE$ 
TEXT 10,110,"5",0,1,1,"DATE2="+STR$(D) 
TEXT 10,210,"5",0,1,1,"DATE3="+STR$(DATE) 
PRINT 1 
EOP 
DEMO 
 

其他參考項目： 

~!C, MONTH, DATE, DAY, HOUR, MINUTE, SECOND  

 WEEK 
 

功能：對 RTC讀出或寫入時間”星期”的值，以 1位數字(1~7)讀出或寫入。 

 

 語法： 

寫入格式：WEEK=3 

讀出格式：A=WEEK 

範圍：1(Sunday)~7(Saturday)  

 

 範例： 

DOWNLOAD "SetWeek.BAS" 
REM *****Set Week Parameter to RTC***** 
WEEK=6 
EOP 
SetWeek 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
 
REM *****Read Week parameter form RTC***** 
WEEK$=STR$(WEEK) 
W=WEEK 
 
REM *****Print***** 
TEXT 10,10,"5",0,1,1,"WEEK1="+WEEK$ 
TEXT 10,110,"5",0,1,1,"WEEK2="+STR$(W) 
TEXT 10,210,"5",0,1,1,"WEEK3="+STR$(WEEK) 
PRINT 1 
EOP 
DEMO 
 

其他參考項目： 

~!C, MONTH, DATE, DAY, HOUR, MINUTE, SECOND  

 HOUR 
 

功能：對 RTC讀出或寫入時間”小時”的值，24小時制以 2位數字(00~23)讀出或寫入。 

 

 語法： 

寫入格式：HOUR=12 

讀出格式：A=HOUR 

範圍：00~23 

 

 範例： 

DOWNLOAD "SetHour.BAS" 
REM *****Set Hour Parameter to RTC***** 
HOUR=11 
EOP 
SetHour 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
 
REM *****Read Hour parameter form RTC***** 
HOUR$=STR$(HOUR) 
H=HOUR 
 
REM *****Print***** 
TEXT 10,10,"5",0,1,1,"HOUR1="+HOUR$ 
TEXT 10,110,"5",0,1,1,"HOUR2="+STR$(H) 
TEXT 10,210,"5",0,1,1,"HOUR3="+STR$(HOUR) 
PRINT 1 
EOP 
DEMO 

 
其他參考項目： 

~!C, MONTH, DATE, DAY, HOUR, MINUTE, SECOND  

 MINUTE 
 

功能：對 RTC讀出或寫入時間”分”的值，以 2位數字(00~59)讀出或寫入。 

 

 語法： 

寫入格式：MINUTE=12 

讀出格式：A=MINUTE 

範圍：00~59 

 

 範例： 

DOWNLOAD "SetMinute.BAS" 
REM *****Set Minute Parameter to RTC***** 
MINUTE=59 
EOP 
SetMinute 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
 
REM *****Read Minute parameter form RTC***** 
MINUTE$=STR$(MINUTE) 
MIN=MINUTE 
 
REM *****Print***** 
TEXT 10,10,"5",0,1,1,"MINUTE1="+MINUTE$ 
TEXT 10,110,"5",0,1,1,"MINUTE2="+STR$(MIN) 
TEXT 10,210,"5",0,1,1,"MINUTE3="+STR$(MINUTE) 
PRINT 1 
EOP 
DEMO 

 
其他參考項目： 

~!C, MONTH, DATE, DAY, HOUR, MINUTE, SECOND  

 SECOND 
 

功能：對 RTC讀出或寫入時間”秒”的值，以 2位數字(00~59)讀出或寫入。 

 

 語法： 

寫入格式：SECOND=12 

讀出格式：A=SECOND 

範圍：00~59 

 

 範例： 

DOWNLOAD "SetSecond.BAS" 
REM *****Set Second Parameter to RTC***** 
SECOND=59 
EOP 
SetSecond 
 
DOWNLOAD "DEMO.BAS" 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 4 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
 
REM *****Read Second parameter form RTC***** 
SECOND$=STR$(SECOND) 
SEC=SECOND 
 
REM *****Print***** 
TEXT 10,10,"5",0,1,1,"SECOND1="+SECOND$ 
TEXT 10,110,"5",0,1,1,"SECOND2="+STR$(SEC) 
TEXT 10,210,"5",0,1,1,"SECOND3="+STR$(SECOND) 
PRINT 1 
EOP 
DEMO 

 
其他參考項目： 

~!C, MONTH, DATE, DAY, HOUR, MINUTE, SECOND  

 @YEAR 
 

功能：對 RTC 讀出或寫入時間”年”的值，以 2 位數字(00~99)讀出或寫入。@YEAR 的使用不

需透過 BASIC程式語法即可直接使用。 

 

 語法： 

寫入格式：@YEAR=”01” 

讀出格式：@YEAR 

範圍：00~99 

 

 範例： 

REM *****Set @YEAR***** 
@YEAR="05" 
 
REM *****Print***** 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 10,10,"5",0,1,1,"@YEAR" 
TEXT 310,10,"5",0,1,1,@YEAR 
PRINT 1 

 
其他參考項目： 

~!C, @MONTH, @DATE, @DAY, @HOUR, @MINUTE, @SECOND 

 @MONTH 
 

功能：對 RTC 讀出或寫入時間”月”的值，以 2 位數字(01~12)讀出或寫入。@MONTH 的使用

不需透過 BASIC 程式語法即可直接使用。 

 

 語法： 

寫入格式：@MONTH=”01” 

讀出格式：@MONTH 

範圍：01~12 

 

 範例： 

REM *****Set @MONTH***** 
@MONTH="12" 
 
REM *****Print***** 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 10,10,"5",0,1,1,"@MONTH" 
TEXT 310,10,"5",0,1,1,@MONTH 
PRINT 1 
 

其他參考項目： 

~!C, @YEAR, @DATE, @DAY, @HOUR, @MINUTE, @SECOND 

 @DATE 
 

功能：對 RTC 讀出或寫入時間”日”的值，以 2 位數字(01~31)讀出或寫入。@DATA 的使用不

需透過 BASIC程式語法即可直接使用。  

 

 語法： 

Write attribute: @DATE=”12” 
Read attribute: @DATE 
Range: 01~31 
 

 範例： 

REM *****Set @DATE***** 
@DATE="31" 
 
REM *****Print***** 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 10,10,"5",0,1,1,"@DATE" 
TEXT 310,10,"5",0,1,1,@DATE 
PRINT 1 
 

其他參考項目： 

~!C, @YEAR, @MONTH, @DAY, @HOUR, @MINUTE, @SECOND 

 @DAY 
 

功能：對 RTC 讀出或寫入時間”星期”的值，以 1 位數字(1~7)讀出或寫入。@DAY 的使用不需

透過 BASIC 程式語法即可直接使用。  

 

 語法： 

寫入格式：@DAY=”3” 

讀出格式：@DAY 

範圍：1(星期日)~7(星期六)  

 

 範例： 

REM *****Set @DAY***** 
@DAY="5" 
 
REM *****Print***** 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 10,10,"5",0,1,1,"@DAY" 
TEXT 310,10,"5",0,1,1,@DAY 
PRINT 1 

 
其他參考項目： 

~!C, @YEAR, @MONTH, @DATE, @HOUR, @MINUTE, @SECOND 

 @HOUR 
 

功能：對 RTC讀出或寫入時間”小時”的值，24小時制以 2位數字(00~23)讀出或寫入。@HOUR

的使用不需透過 BASIC程式語法即可直接使用。  

 

 語法： 

寫入格式：@HOUR =”12” 

讀出格式：@HOUR 

範圍：00~23 

 

 範例： 

REM *****Set @HOUR***** 
@HOUR="23" 
 
REM *****Print***** 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 10,10,"5",0,1,1,"@HOUR" 
TEXT 310,10,"5",0,1,1,@HOUR 
PRINT 1 

 
其他參考項目： 

~!C, @YEAR, @MONTH, @DATE, @DAY, @MINUTE, @SECOND 

 @MINUTE 
 

功能：對 RTC 讀出或寫入時間”分”的值，以 2 位數字(00~59)讀出或寫入。@MINUTE 的使用

不需透過 BASIC 程式語法即可直接使用。  

 

 語法： 

寫入格式：@MINUTE =”12” 

讀出格式：@MINUTE 

範圍：00~59 

 

 範例： 

REM *****Set @MINUTE***** 
@MINUTE="59" 
 
REM *****Print***** 
SIZE 3,3 
GAP 0.08,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 10,10,"5",0,1,1,"@MINUTE" 
TEXT 310,10,"5",0,1,1,@MINUTE 
PRINT 1 
 

其他參考項目： 

~!C, @YEAR, @MONTH, @DATE, @DAY, @HOUR, @SECOND 

 @SECOND 
 

功能：對 RTC讀出或寫入時間”秒”的值，以 2位數字(00~59)讀出或寫入。@SECOND的使用

不需透過 BASIC 程式語法即可直接使用。  

 

 語法： 

寫入格式：@SECOND=”12” 

讀出格式：@SECOND 

範圍：00~59 

 

 範例： 

REM *****Set @SECOND***** 
@SECOND="59" 
 
REM *****Print***** 
SIZE 3,3 
GAP 0,0 
DENSITY 8 
SPEED 6 
DIRECTION 0 
REFERENCE 0,0 
SET CUTTER OFF 
SET PEEL OFF 
CLS 
TEXT 10,10,"5",0,1,1,"@SECOND" 
TEXT 310,10,"5",0,1,1,@SECOND 
PRINT 1 

 
其他參考項目： 

~!C, @YEAR, @MONTH, @DATE, @DAY, @HOUR, @MINUTE 



 

 

 
 

 
 
 
 
 
 
 
 
 
 
 
 

 


